
Pow is a robust, modular, and extendable authentication and user management solution for Phoenix and Plug-based apps.

  
    
  
  Features


	User registration
	Session based authorization
	Per Endpoint/Plug configuration
	API token authorization
	Mnesia cache with automatic cluster healing
	Multitenancy
	User roles
	Extendable
	I18n
	And more


  
    
  
  Installation


Add Pow to your list of dependencies in mix.exs:
defp deps do
  [
    # ...
    {:pow, "~> 1.0.39"}
  ]
end
Run mix deps.get to install it.

  
    
  
  Getting started



  
    
  
  Phoenix app


Umbrella project: Check out the umbrella project guide.
Install the necessary files:
mix pow.install

This will add the following files to your app:
LIB_PATH/users/user.ex
PRIV_PATH/repo/migrations/TIMESTAMP_create_users.ex

And also update the following files:
config/config.exs
WEB_PATH/endpoint.ex
WEB_PATH/router.ex

Run migrations with mix setup, start the server with mix phx.server, and you can now visit http://localhost:4000/registration/new to create a user.

  
    
  
  Modify templates


By default, Pow exposes as few files as possible.
If you wish to modify the templates, you can generate them using:
mix pow.phoenix.gen.templates

This will also add web_module: MyAppWeb to the configuration in config/config.exs.

  
    
  
  Extensions


Pow is made so it's easy to extend the functionality with your own complimentary library. The following extensions are included in this library:
	PowResetPassword
	PowEmailConfirmation
	PowPersistentSession
	PowInvitation

Check out the "Other libraries" section for other extensions.

  
    
  
  Add extensions support


To keep it easy to understand and configure Pow, you'll have to enable the extensions yourself.
Let's install the PowResetPassword and PowEmailConfirmation extensions.
First, install extension migrations by running:
mix pow.extension.ecto.gen.migrations --extension PowResetPassword --extension PowEmailConfirmation

Then run the migrations with mix ecto.migrate. Now, update config/config.ex with the :extensions and :controller_callbacks key:
config :my_app, :pow,
  user: MyApp.Users.User,
  repo: MyApp.Repo,
  extensions: [PowResetPassword, PowEmailConfirmation],
  controller_callbacks: Pow.Extension.Phoenix.ControllerCallbacks
Update LIB_PATH/users/user.ex with the extensions:
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema
  use Pow.Extension.Ecto.Schema,
    extensions: [PowResetPassword, PowEmailConfirmation]

  # ...

  def changeset(user_or_changeset, attrs) do
    user_or_changeset
    |> pow_changeset(attrs)
    |> pow_extension_changeset(attrs)
  end
end
Add Pow extension routes to WEB_PATH/router.ex:
defmodule MyAppWeb.Router do
  use MyAppWeb, :router
  use Pow.Phoenix.Router
  use Pow.Extension.Phoenix.Router,
    extensions: [PowResetPassword, PowEmailConfirmation]

  # ...

  scope "/" do
    pipe_through :browser

    pow_routes()
    pow_extension_routes()
  end

  # ...
end
Modify extension templates
Templates for extensions can be generated with:
mix pow.extension.phoenix.gen.templates --extension PowResetPassword --extension PowEmailConfirmation

Please follow the instructions in "Modify templates" to ensure that your custom templates will be used.

  
    
  
  Mailer support


Many extensions require a mailer to have been set up. Let's create a mailer mock module in  WEB_PATH/mails/pow/mailer.ex:
defmodule MyAppWeb.Pow.Mailer do
  use Pow.Phoenix.Mailer
  require Logger

  def cast(%{user: user, subject: subject, text: text, html: html, assigns: _assigns}) do
    # Build email struct to be used in `process/1`

    %{to: user.email, subject: subject, text: text, html: html}
  end

  def process(email) do
    # Send email

    Logger.debug("E-mail sent: #{inspect email}")
  end
end
Update config/config.ex with :mailer_backend key:
config :my_app, :pow,
  # ...
  mailer_backend: MyAppWeb.Pow.Mailer
This mailer module will only output the mail to your log, so you can e.g. try out the reset password and email confirmation links. You should integrate the Pow mailer with your actual mailer system. For Swoosh or Bamboo integration, check out the Configuring mailer guide.
Modify mailer templates
Generate the template files:
mix pow.extension.phoenix.mailer.gen.templates --extension PowResetPassword --extension PowEmailConfirmation

This will generate template files in the WEB_PATH/mails/ directory. This will also add the necessary mail/0 macro to WEB_PATH/my_app_web.ex and update the pow config with web_mailer_module: MyAppWeb.

  
    
  
  Configuration


Pow is built to be modular, and easy to configure. The configuration is passed to function calls as well as plug options, and they will take priority over any environment configuration. It's ideal in case you got an umbrella app with multiple separate user domains.
The easiest way to use Pow with Phoenix is to use a :otp_app in function calls and set the app environment configuration. It will keep a persistent fallback configuration that you configure in one place.

  
    
  
  Module groups


Pow has three main groups of modules that each can be used individually, or in conjunction with each other:
Pow.Plug
This group will handle the plug connection. The configuration will be assigned to conn.private[:pow_config] and passed through the controller to the users' context module. The Plug module has functions to authenticate, create, update, and delete users, and will generate/renew the session automatically.
Pow.Ecto
This group contains all modules related to the Ecto based user schema and context. By default, Pow will use the Pow.Ecto.Context module to authenticate, create, update and delete users with lookups to the database. However, it's straightforward to extend or write your custom user context. You can do this by setting the :users_context configuration key.
Pow.Phoenix
This group contains the controllers and templates for Phoenix. You only need to set the (session) plug in endpoint.ex and add the routes to router.ex. Templates are not generated by default, instead, the compiled templates in Pow are used. You can generate the templates used by running mix pow.phoenix.gen.templates. You can also customize flash messages and callback routes by creating your own using :messages_backend and :routes_backend.
The registration and session controllers can be changed with your customized versions too, but since the routes are built on compile time, you'll have to set them up in router.ex with :pow namespace. For minor pre/post-processing of requests, you can use the :controller_callbacks option. It exists to make it easier to modify flow with extensions (e.g., send a confirmation email upon user registration).

  
    
  
  Pow.Extension


This module helps build extensions for Pow. There're three extension mix tasks to generate Ecto migrations and phoenix templates.
mix pow.extension.ecto.gen.migrations

mix pow.extension.phoenix.gen.templates

mix pow.extension.phoenix.mailer.gen.templates


  
    
  
  Authorization plug


Pow ships with a session plug module. You can easily switch it out with a different one. As an example, here's how you do that with Phoenix.Token:
defmodule MyAppWeb.Pow.Plug do
  use Pow.Plug.Base

  @session_key :pow_user_token
  @salt "user salt"
  @max_age 86400

  def fetch(conn, config) do
    conn  = Plug.Conn.fetch_session(conn)
    token = Plug.Conn.get_session(conn, @session_key)

    MyAppWeb.Endpoint
    |> Phoenix.Token.verify(@salt, token, max_age: @max_age)
    |> maybe_load_user(conn)
  end

  defp maybe_load_user({:ok, user_id}, conn), do: {conn, MyApp.Repo.get(User, user_id)}
  defp maybe_load_user({:error, _any}, conn), do: {conn, nil}

  def create(conn, user, config) do
    token = Phoenix.Token.sign(MyAppWeb.Endpoint, @salt, user.id)
    conn  =
      conn
      |> Plug.Conn.fetch_session()
      |> Plug.Conn.put_session(@session_key, token)

    {conn, user}
  end

  def delete(conn, config) do
    conn
    |> Plug.Conn.fetch_session()
    |> Plug.Conn.delete_session(@session_key)
  end
end

defmodule MyAppWeb.Endpoint do
  # ...

  plug MyAppWeb.Pow.Plug, otp_app: :my_app
end

  
    
  
  Ecto changeset


The user module has a fallback changeset/2 function. If you want to add custom validations, you can use the pow_changeset/2 function like so:
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema

  schema "users" do
    field :custom, :string

    pow_user_fields()

    timestamps()
  end

  def changeset(user_or_changeset, attrs) do
    user_or_changeset
    |> pow_changeset(attrs)
    |> Ecto.Changeset.cast(attrs, [:custom])
    |> Ecto.Changeset.validate_required([:custom])
  end
end

  
    
  
  Phoenix controllers


Controllers in Pow are very slim and consists of just one Pow.Plug function call with response functions. If you wish to change the flow of the Pow.Phoenix.RegistrationController and Pow.Phoenix.SessionController, the best way is to create your own and modify router.ex.
However, to make it easier to integrate extension, you can add callbacks to the controllers that do some light pre/post-processing of the request:
defmodule MyCustomExtension.Phoenix.ControllerCallbacks do
  use Pow.Extension.Phoenix.ControllerCallbacks.Base

  def before_respond(Pow.Phoenix.RegistrationController, :create, {:ok, user, conn}, _config) do
    # send email

    {:ok, user, conn}
  end
end
You can add functions for before_process/4 (before the action happens) and before_respond/4 (before parsing the results from the action).
Testing with authenticated users
To test with authenticated users in your controller tests, you just have to assign the user to the conn in your setup callback:
setup %{conn: conn} do
  user = %User{email: "test@example.com"}
  conn = Pow.Plug.assign_current_user(conn, user, otp_app: :my_app)

  {:ok, conn: conn}
end

  
    
  
  I18n


All templates can be generated and modified to use your Gettext module.
For flash messages, you can create the following module:
defmodule MyAppWeb.Pow.Messages do
  use Pow.Phoenix.Messages
  use Pow.Extension.Phoenix.Messages,
    extensions: [PowResetPassword]

  import MyAppWeb.Gettext

  def user_not_authenticated(_conn), do: gettext("You need to sign in to see this page.")

  # Message functions for extensions has to be prepended with the snake cased
  # extension name. So the `email_has_been_sent/1` function from
  # `PowResetPassword` is written as `pow_reset_password_email_has_been_sent/1`
  # in your messages module.
  def pow_reset_password_email_has_been_sent(_conn), do: gettext("An email with reset instructions has been sent to you. Please check your inbox.")
end
Add messages_backend: MyAppWeb.Pow.Messages to your configuration. You can find all the messages in Pow.Phoenix.Messages and [Pow Extension].Phoenix.Messages.

  
    
  
  Callback routes


You can customize callback routes by creating the following module:
defmodule MyAppWeb.Pow.Routes do
  use Pow.Phoenix.Routes
  use MyAppWeb, :verified_routes

  def after_sign_in_path(conn), do: ~p"/home"
end
Add routes_backend: MyAppWeb.Pow.Routes to your configuration. You can find all the routes in Pow.Phoenix.Routes.

  
    
  
  Password hashing function


Pow.Ecto.Schema.Password is used to hash and verify with Pbkdf2. It's highly recommended to lower the iterations count in your test environment to speed up your tests:
config :pow, Pow.Ecto.Schema.Password, iterations: 1
You can change the password hashing function easily. For example, this is how you use comeonin with Argon2:
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema,
    password_hash_verify: {&Argon2.hash_pwd_salt/1,
                           &Argon2.verify_pass/2}

  # ...
end

  
    
  
  Current user and sign out link


You can use Pow.Plug.current_user/1 to fetch the current user from the connection.
This can be used to show the sign in or sign out links in your Phoenix template:
<.link :if={Pow.Plug.current_user(@conn)} href={~p"/session"} method="delete">Sign out</.link>
<.link :if={is_nil Pow.Plug.current_user(@conn)} navigate={~p"/registration/new"}>Registration</.link>
<.link :if={is_nil Pow.Plug.current_user(@conn)} navigate={~p"/session/new"}>Sign In</.link>
The current user can also be fetched by using the template assigns set in the configuration with :current_user_assigns_key (defaults to @current_user).

  
    
  
  Plugs



  
    
  
  Pow.Plug.Session


Enables session-based authorization. The user struct will be collected from a cache store through a GenServer using a unique token generated for the session. The token will be reset every time the authorization level changes (handled by Pow.Plug) or after a certain interval (default 15 minutes).
The user struct fetched can be out of sync with the database if the row in the database is updated by actions outside Pow. In this case, it's recommended to add a plug that reloads the user struct and reassigns it to the connection.
Custom metadata can be set for the session by assigning a private :pow_session_metadata key in the conn. Read the Pow.Plug.Session module docs for more details.
Cache store
By default Pow.Store.Backend.EtsCache is started automatically and can be used in development and test environment.
For a production environment, you should use a distributed, persistent cache store. Pow makes this easy with Pow.Store.Backend.MnesiaCache. To start MnesiaCache in your Phoenix app, add it to your application.ex supervisor:
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      MyApp.Repo,
      MyAppWeb.Endpoint,
      Pow.Store.Backend.MnesiaCache
      # # Or in a distributed system:
      # {Pow.Store.Backend.MnesiaCache, extra_db_nodes: {Node, :list, []}},
      # Pow.Store.Backend.MnesiaCache.Unsplit # Recover from netsplit
    ]

    opts = [strategy: :one_for_one, name: MyAppWeb.Supervisor]
    Supervisor.start_link(children, opts)
  end

  # ...
end
Update config/config.ex with :cache_store_backend key:
config :my_app, :pow,
  # ...
  cache_store_backend: Pow.Store.Backend.MnesiaCache
Remember to add :mnesia to your :extra_applications so it'll be available for your release build. Mnesia will write files to the current working directory. The path can be changed with config :mnesia, dir: ~c"/path/to/dir".
The MnesiaCache requires write access. If you've got a read-only file system you should take a look at the Redis cache backend store guide.

  
    
  
  Pow.Plug.RequireAuthenticated


Will halt connection if no current user is not present in assigns. Expects an :error_handler option.

  
    
  
  Pow.Plug.RequireNotAuthenticated


Will halt connection if a current user is present in assigns. Expects an :error_handler option.

  
    
  
  Pow security practices


See security practices.

  
    
  
  Other libraries


PowAssent - Multi-provider support for Pow with strategies for Twitter, Github, Google, Facebook and more

  
    
  
  Contributing


Please read CONTRIBUTING.md.

  
    
  
  LICENSE


(The MIT License)
Copyright (c) 2018-2019 Dan Schultzer & the Contributors
Permission is hereby granted, free of charge, to any person obtaining a copy of this software and associated documentation files (the 'Software'), to deal in the Software without restriction, including without limitation the rights to use, copy, modify, merge, publish, distribute, sublicense, and/or sell copies of the Software, and to permit persons to whom the Software is furnished to do so, subject to the following conditions:
The above copyright notice and this permission notice shall be included in all copies or substantial portions of the Software.
THE SOFTWARE IS PROVIDED 'AS IS', WITHOUT WARRANTY OF ANY KIND, EXPRESS OR IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY, FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM, OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE SOFTWARE.


  

    
Contributing to Pow
    

All contributions to Pow are welcome!
Postgres is required to test locally. The test helper will automatically set up the database for you when you run the tests. Run mix credo for static code analysing.

  
    
  
  Postgres setup


The test environment has minimal database configuration which means it'll by default use the current username in the OS. If no such role exists (common with older postgres setup where only postgres role exists), you can easily create the superuser user by running:
createuser -U postgres -s $(whoami)


  
    
  
  Mocks


As the default ETS cache store backend works asynchronously, a synchronous ETS cache store backend (Pow.Test.EtsCacheMock) is used instead.
Only Ecto modules are tested against the database. Plug and Phoenix modules uses Pow.Test.ContextMock.

  
    
  
  Extension test support


Due to compile-time configuration of Phoenix modules and User schema modules, several modules are dynamically generated with Pow.Test.ExtensionMocks.

  
    
  
  Releases


Releases to hex are automatically handled in Travis CI, when a tag is pushed on Github (and a new version has been set in mix.exs).


  

    
Changelog
    


  
    
  
  v1.0.39 (2025-01-11)


Now requires Elixir 1.14+.

  
    
  
  Bug fixes


	[Pow.Extension.Ecto.Schema] Fixed deprecation warning in Elixir 1.18
	[Mix.Pow.Ecto.Migration] Fixed compilation warning in Elixir 1.18
	[Pow.Ecto.Schema] Fixed issues caused by changes in Ecto 3.12.0


  
    
  
  v1.0.38 (2024-04-11)



  
    
  
  Bug fixes


	[Pow.Ecto.Schema.Changeset] Fixed issue with schemas having a :data field not being handled correctly in changeset functions


  
    
  
  v1.0.37 (2024-03-03)



  
    
  
  Bug fixes


	[Pow.Phoenix.Template] Now sets the function annotation in caller environment to prevent compile error when :debug_heex_annotations is set to true


  
    
  
  v1.0.36 (2024-02-07)



  
    
  
  Bug fixes


	[Pow.Phoenix.ViewHelpers] Now ensures format for layouts in Phoenix 1.7 is conformed to prevent conflicting layouts found warnings


  
    
  
  v1.0.35 (2024-01-04)


Handles Elixir 1.16 deprecations.

  
    
  
  Enhancements


	[Pow.Ecto.Schema.Changeset] Now handles MFA for :password_hash_verify
	[Pow.Ecto.Schema.Changeset] Now handles MFA for :email_validator


  
    
  
  Deprecations


	[Pow.Ecto.Schema.Changeset] Deprecated :password_hash_methods in favor of :password_hash_verify


  
    
  
  v1.0.34 (2023-09-18)


Note: This release contains an important security fix. It is recommended to update  immediately if you are using the Pow.Store.Backend.MnesiaCache.

  
    
  
  Bug fixes


	[Pow.Store.Backend.MnesiaCache] Fixed bug where expired cached keys are not invalidated on startup


  
    
  
  v1.0.33 (2023-09-05)



  
    
  
  Bug fixes


	[Pow.Phoenix.Controller] Fixed regression bug with Phoenix.View and :namespace option
	[Pow.Phoenix.ViewHelpers] Now falls back to view named modules to prevent upgrade issues


  
    
  
  v1.0.32 (2023-08-30)


Removed deprecation warnings for Elixir 1.15.

  
    
  
  Bug fixes


	[Pow.Phoenix.Controller] Now uses Phoenix.View when it's available with :namespace option to prevent upgrade issues


  
    
  
  v1.0.31 (2023-06-09)



  
    
  
  Bug fixes


	[Pow.Phoenix.Mailer.Mail] Now renders html and text correctly in fallback mode for deprecated MailView


  
    
  
  v1.0.30 (2023-04-28)



  
    
  
  Bug fixes


	[Pow] Ensure the dependency is loaded before matching version in Pow.dependency_vsn_match?/2


  
    
  
  v1.0.29 (2023-03-21)


This version is updated to work with Phoenix 1.7, in particular using the new template components structure. All views have been removed, and Pow no longer requires the phoenix_view dependency.
Instead of pow_mailer_layout: {MyAppWeb.LayoutView, :email}) you should use pow_mailer_layouts: [html: {MyAppWeb.Layouts, :email}, text: {MyAppWeb.Layouts, :email_text}] in conn.private.
Now requires Elixir 1.12+.

  
    
  
  Enhancements


	[Pow.Phoenix.Mailer.Mail] Revamped to conform to template component structure of Phoenix 1.7
	[Pow.Phoenix.Template] Now renders Tailwind based template components on Phoenix 1.7


  
    
  
  Deprecations


	[Pow.Phoenix.Mailer.Mail] Deprecated :pow_mailer_layout in favor of :pow_mailer_layouts
	[Mix.Pow.Phoenix.Mailer] Removed Mix.Pow.Phoenix.Mailer.create_view_file/5
	[Mix.Pow.Phoenix.Mailer] Removed Mix.Pow.Phoenix.Mailer.create_templates/4
	[Mix.Pow.Phoenix] Removed Mix.Pow.Phoenix.create_view_file/4


  
    
  
  v1.0.28 (2023-03-17)



  
    
  
  Enhancements


	[Mix.Pow] Mix.Pow.parse_options/3 now merges option defaults with :otp_app, :generators configuration
	[Mix.Pow.Mix.Tasks.Pow.Phoenix.Mailer.Gen.Templates] Now injects config/config.exs and WEB_PATH/WEB_APP.ex
	[Mix.Pow.Mix.Tasks.Pow.Phoenix.Gen.Templates] Now injects config/config.exs
	[Mix.Tasks.Pow.Phoenix.Install] Now injects config/config.exs, WEB_PATH/endpoint.ex, and WEB_PATH/router.ex
	[Pow.Phoenix.Router] Updated to support Phoenix 1.7 breaking changes
	[Pow.Phoenix.Template] Updated to support Phoenix 1.7 verified routes
	[Pow.Phoenix.Routes] Updated to support Phoenix 1.7 verified routes
	[Pow.Phoenix.ViewHelpers] Updated to handle Phoenix 1.7 components layout


  
    
  
  Bug fixes


	:phoenix removed from the compilers


  
    
  
  Documentation


	Updated api guide to correctly return updated conn for delete calls


  
    
  
  v1.0.27 (2022-04-27)


Now supports ecto_sql 3.8.x and requires Elixir 1.11+.

  
    
  
  Enhancements


	[Pow.Ecto.Schema] has been refactored to conform the @pow_fields and @pow_assocs attributes with separate migration options


  
    
  
  v1.0.26 (2021-11-06)



  
    
  
  Enhancemnets


	[Pow.Store.Backend.MnesiaCache.Unsplit] The unsplit module will now initialize the Mnesia cluster when nodes are connected lazily by resetting the Mnesia schema


  
    
  
  Bug fixes


	[Pow.Store.Backend.MnesiaCache] Now properly handles Mnesia application start errors


  
    
  
  Documentation


	Updated api guide to use Plug.Conn.register_before_send/2 for token writes


  
    
  
  v1.0.25 (2021-09-26)


Now supports Phoenix 1.6.x, and phoenix_html 3.x.x.

  
    
  
  Enhancements


	[Pow.Ecto.Schema.Fields] The :password_hash, :current_password, and :password fields now have redact: true option set
	[Pow.Phoenix.Controller] Pow.Phoenix.Controller.action/3 now properly handles {:halt, conn} returned in the before_process callback
	[Pow.Store.Backend.EtsCache] Now does synchronous writes unless writes: :async is passed in config options
	[Pow.Store.Backend.MnesiaCache] Now does synchronous writes unless writes: :async is passed in config options


  
    
  
  Bug fixes


	[Pow.Operations] Pow.Operations.fetch_primary_key_values/2 now ensures that module exists and is loaded before deriving primary keys


  
    
  
  Documentation


	Updated redis guide to use synchronous writes unless writes: :async is passed in config options
	Updated redis guide to use optimized lookups with sorted keys


  
    
  
  v1.0.24 (2021-05-27)



  
    
  
  Enhancements


	[Pow.Store.Backend.MnesiaCache] Now accepts extra_db_nodes: {module, function, arguments} to fetch nodes when MnesiaCache starts up
	[PowEmailConfirmation.Phoenix.Messages] Added PowEmailConfirmation.Phoenix.Messages.invalid_token/1
	[Pow.Store.CredentialsCache] Now outputs an IO warning when a :ttl longer than 30 minutes is used


  
    
  
  Bug fixes


	[Pow.Store.Backend.MnesiaCache] Now handles initialization errors


  
    
  
  v1.0.23 (2021-03-22)



  
    
  
  Enhancements


	[Pow.Ecto.Context] No longer automatically reloads the struct after insert or update
	[PowInvitation.Ecto.Schema] Added PowInvitation.Ecto.Schema.invitation_token_changeset/1
	[PowInvitation.Ecto.Schema] Added PowInvitation.Ecto.Schema.invited_by_changeset/2
	[Pow.Ecto.Schema.Password.Pbkdf2] Now uses :crypto.mac/4 if available to support OTP 24
	[PowEmailConfirmation.Phoenix.ControllerCallbacks] Now returns :info instead of :error message for when the user has to confirm their email


  
    
  
  Bug fixes


	[Pow.Store.Backend.MnesiaCache] No longer triggers Elixir 1.11 dependency warnings


  
    
  
  v1.0.22 (2021-01-27)


This release introduces a deprecation for the default API guide implementation. Please check migration section below.

  
    
  
  Enhancements


	[PowPersistentSession.Plug.Cookie] Now stores the user struct instead of clauses
	[PowPersistentSession.Plug.Base] Now includes :pow_config in the store config
	[PowResetPassword.Plug] Now includes :pow_config in the store config
	[Pow.Plug.Base] Now includes :pow_config in the store config
	[Pow.Operations] Added Pow.Operations.reload/2 to reload structs
	[PowPersistentSession.Store.PersistentSessionCache] Update PowPersistentSession.Store.PersistentSessionCache.get/2 to reload the user using Pow.Operations.reload/2
	[Pow.Store.CredentialsCache] Now support reload: true configuration so once fetched from the cache the user object will be reloaded through the context module


  
    
  
  Documentation


	Updated the API guide as it's no longer necessary to load the user struct


  
    
  
  Migration


If you've used an API setup for previous version, you'll see the warning PowPersistentSession.Store.PersistentSessionCache.get/2 call without `:pow_config` in second argument is deprecated, refer to the API guide.. It's recommended to replace your APIAuthPlug with the updated version in the API guide.
The larger refactor of cache setup in Pow v1.0.22 means that user struct is always expected to be passed in and returned by the stores, so it is no longer necessary to load the user in the API plug. The PowPersistentSession.Store.PersistentSessionCache has fallback logic to handle the deprecated clauses keyword list, and will load the user correctly.

  
    
  
  v1.0.21 (2020-09-13)



  
    
  
  Enhancements


	[Pow.Plug.Base] Will now use the existing :pow_config in the conn when no plug options has been set
	[PowInvitation.Phoenix.InvitationController] Fixed bug where user was incorrectly redirected to the show action with unsigned token when user struct has no e-mail
	[Pow.Ecto.Schema] Now only emits warning for primitive Ecto types


  
    
  
  Bug fixes


	[PowEmailConfirmation.Ecto.Schema] PowEmailConfirmation.Ecto.Schema.changeset/3 no longer sets the email to the unconfirmed email when the same email change is set twice
	[Pow.Extension.Phoenix.Messages] Fixed fallback message dializer warning
	[Pow.Ecto.Context] Fixed bug where the macro didn't add :users_context to the Pow config in the module resulting in Pow.Ecto.Context.get_by/2 being called instead of get_by/1 in the custom context
	[Pow.Ecto.Schema.Changeset] The Pow.Ecto.Schema.Changeset.validate_email/1 method has been improved per specifications to support wider unicode support, fully-qualified domain validation, and comments


  
    
  
  v1.0.20 (2020-04-22)


Now supports Phoenix 1.5, and requires Elixir 1.7 or higher.

  
    
  
  Enhancements


	[Mix.Tasks.Pow.Extension.Phoenix.Gen.Templates] mix pow.extension.phoenix.gen.templates now dynamically loads template list from the extension base module
	[PowResetPassword.Plug] PowResetPassword.Plug.load_user_by_token/2 now sets a :pow_reset_password_decoded_token key in conn.private that will be used in PowResetPassword.Plug.update_user_password/2


  
    
  
  v1.0.19 (2020-03-13)


Warning: This release will now sign and verify all tokens, causing previous tokens to no longer work. Any sessions and persistent sessions will be invalidated.

  
    
  
  Enhancements


	[Pow.Plug.Session] Now sets a global lock when renewing the session
	[PowPersistentSession.Plug.Cookie] Now sets a global lock when authenticating the user
	[PowEmailConfirmation.Plug] Added PowEmailConfirmation.Plug.sign_confirmation_token/2 to sign the email_confirmation_token to prevent timing attacks
	[PowEmailConfirmation.Plug] Added PowEmailConfirmation.Plug.load_user_by_token/2 to verify the signed email_confirmation_token to prevent timing attacks
	[PowEmailConfirmation.Plug] Added PowEmailConfirmation.Plug.confirm_email/2 with map as second argument
	[PowInvitation.Plug] Added PowInvitation.Plug.sign_invitation_token/2 to sign the invitation_token
	[PowInvitation.Plug] Added PowInvitation.Plug.load_invited_user_by_token/2 to verify the signed invitation_token to prevent timing attacks
	[PowResetPassword.Plug] Changed PowResetPassword.Plug.create_reset_token/2 to sign the :token
	[PowResetPassword.Plug] Added PowResetPassword.Plug.load_user_by_token/2 to verify the signed token to prevent timing attacks
	[PowResetPassword.Plug] Changed PowResetPassword.Plug.update_user_password/2 so it decodes the signed token
	[PowPersistentSession.Plug.Cookie] Now uses signed tokens to prevent timing attacks
	[Pow.Plug.Session] Now uses signed session ID's to prevent timing attacks
	[Pow.Plug] Added Pow.Plug.sign_token/4 to sign tokens
	[Pow.Plug] Added Pow.Plug.verify_token/4 to decode and verify signed tokens
	[Pow.Plug.MessageVerifier] Added Pow.Plug.MessageVerifier module to sign and verify messages
	[PowEmailConfirmation.Ecto.Context] Added PowEmailConfirmation.Ecto.Context.confirm_email/3
	[PowEmailConfirmation.Ecto.Schema] Added confirm_email_changeset/2 and pow_confirm_email_changeset/2 to the macro
	[PowEmailConfirmation.Ecto.Schema] Added PowEmailConfirmation.Ecto.Schema.confirm_email_changeset/2
	[PowInvitation.Ecto.Schema] Added accept_invitation_changeset/2 and pow_accept_invitation_changeset/2 to the macro
	[PowResetPassword.Ecto.Schema] Added reset_password_changeset/2 and pow_reset_password_changeset/2 to the macro
	[Pow.Ecto.Schema] Now emits a warning instead of raising error with missing fields/associations


  
    
  
  Deprecations


	[PowEmailConfirmation.Plug] PowEmailConfirmation.Plug.confirm_email/2 with token param as second argument has been deprecated in favor of PowEmailConfirmation.Plug.load_user_by_token/2, and PowEmailConfirmation.Plug.confirm_email/2 with map as second argument
	[PowInvitation.Plug] PowInvitation.Plug.invited_user_from_token/2 has been deprecated in favor of PowInvitation.Plug.load_invited_user_by_token/2
	[PowInvitation.Plug] PowInvitation.Plug.assign_invited_user/2 has been deprecated
	[PowResetPassword.Plug] PowResetPassword.Plug.user_from_token/2 has been deprecated in favor of PowResetPassword.Plug.load_user_by_token/2
	[PowResetPassword.Plug] PowResetPassword.Plug.assign_reset_password_user/2 has been deprecated
	[PowEmailConfirmation.Ecto.Context] PowEmailConfirmation.Ecto.Context.confirm_email/2 deprecated in favor of PowEmailConfirmation.Ecto.Context.confirm_email/3
	[PowEmailConfirmation.Ecto.Schema] PowEmailConfirmation.Ecto.Schema.confirm_email_changeset/1 deprecated in favor of PowEmailConfirmation.Ecto.Schema.confirm_email_changeset/2


  
    
  
  Documentation


	Updated the API guide with signed tokens


  
    
  
  v1.0.18 (2020-02-14)



  
    
  
  Bug fixes


	[Pow.Phoenix.Routes] Fixed bug where callback route methods is not using the overridden method
	[PowPersistentSession.Plug.Cookie] PowPersistentSession.Plug.Cookie.delete/2 now correctly pulls token during :before_send callback
	[Pow.Plug.Session] Pow.Plug.Session.delete/2 now correctly pulls session id during :before_send callback so PowEmailConfirmation will remove set session


  
    
  
  v1.0.17 (2020-02-04)



  
    
  
  Enhancements


	[Pow.Ecto.Context] Calls to Pow.Ecto.Context.get_by/2 replaced with Pow.Operations.get_by/2 so custom users context module can be used. The following methods has been updated:	Pow.Ecto.Context.authenticate/2
	PowEmailConfirmation.Ecto.Context.get_by_confirmation_token/2
	PowInvitation.Ecto.Context.get_by_invitation_token/2
	PowResetPassword.Ecto.Context.get_by_email/2


	[Pow.Ecto.Schema.Changeset] Pow.Ecto.Schema.Changeset.confirm_password_changeset/3 now adds the default Ecto.Changeset.validate_confirmation/3 error instead of the previous not same as password error
	[Pow.Ecto.Schema.Changeset] Pow.Ecto.Schema.Changeset.confirm_password_changeset/3 now uses the Ecto.Changeset.validate_confirmation/3 for validation and expects :password_confirmation instead of :confirm_password in params
	[Pow.Ecto.Schema.Changeset] Pow.Ecto.Schema.Changeset.new_password_changeset/3 now only requires the :password_hash if there have been no previous errors set in the changeset
	[Pow.Ecto.Schema] No longer adds :confirm_password virtual field
	[Pow.Ecto.Schema] Now has an @after_compile callback that ensures all required fields has been defined
	[PowInvitation.Phoenix.InvitationView] Now renders :password_confirmation field instead of :confirm_password
	[PowResetPassword.Phoenix.ResetPasswordView] Now renders :password_confirmation field instead of :confirm_password
	[Pow.Phoenix.RegistrationView] Now renders :password_confirmation field instead of :confirm_password
	[PowEmailConfirmation.Ecto.Schema] No longer validates if :email has been taken before setting :unconfirmed_email
	[PowEmailConfirmation.Phoenix.ControllerCallbacks] Now prevents user enumeration for PowInvitation.Phoenix.InvitationController.create/2
	[PowPersistentSession.Plug.Cookie] Changed default cookie name to persistent_session
	[PowPersistentSession.Plug.Cookie] Removed renewal of cookie as the token will always expire
	[PowPersistentSession.Plug.Cookie] No longer expires invalid cookies
	[Pow.Operations] Added Pow.Operations.fetch_primary_key_values/2
	[PowPersistentSession.Plug.Base] Now registers :before_send callbacks
	[PowPersistentSession.Plug.Cookie] Now updates cookie and backend store in :before_send callback
	[Pow.Plug.Base] Now registers :before_send callbacks
	[Pow.Plug.Session] Now updates plug session and backend store in  :before_send callback
	[Pow.Plug] Added Pow.Plug.create/3
	[Pow.Plug] Added Pow.Plug.delete/2


  
    
  
  Removed


	[PowResetPassword.Phoenix.ResetPasswordController] Will no longer prevent information leak by checking if PowEmailConfirmation or registration routes are enabled; instead it'll by default prevent user enumeration, but can be disabled if pow_prevent_user_enumeration: false is set in conn.private


  
    
  
  Bug fixes


	[PowPersistentSession.Plug.Base] With custom :persistent_session_store now falls back to :cache_store_backend configuration option
	[PowResetPassword.Plug] With custom :reset_password_token_store now falls back to :cache_store_backend configuration option
	[Pow.Plug.Base] With custom :credentials_cache_store now falls back to :cache_store_backend configuration option


  
    
  
  Deprecations


	[Pow.Ecto.Changeset] Pow.Ecto.Schema.Changeset.confirm_password_changeset/3 has deprecated use of :confirm_password in params in favor of :password_confirmation
	[Pow.Plug.Session] :session_store option has been renamed to :credentials_cache_store
	[Pow.Plug] Pow.Plug.clear_authenticated_user/1 deprecated in favor of Pow.Plug.delete/1


  
    
  
  v1.0.16 (2020-01-07)


Note: This release contains an important security fix.

  
    
  
  Enhancements


	[PowPersistentSession.Plug.Cookie] Now supports :persistent_session_cookie_opts to customize any options that will be passed on to Plug.Conn.put_resp_cookie/4
	[PowResetPassword.Phoenix.ResetPasswordController] Now uses PowResetPassword.Phoenix.Messages.maybe_email_has_been_sent/1 with a generic response that tells the user the email has been sent only if an account was found
	[PowResetPassword.Phoenix.ResetPasswordController] When a user doesn't exist will now return success message if PowEmailConfirmation extension is enabled
	[PowResetPassword.Phoenix.Messages] Added PowResetPassword.Phoenix.Messages.maybe_email_has_been_sent/1 and let PowResetPassword.Phoenix.Messages.email_has_been_sent/1 fall back to it
	[PowEmailConfirmation.Phoenix.ControllerCallbacks] When a user tries to sign up and the email has already been taken the default e-mail confirmation required message will be shown
	[Pow.Plug.Session] Now renews the Plug session each time the Pow session is created or rolled


  
    
  
  Bug fixes


	[Pow.Ecto.Schema.Changeset] Fixed bug where Pow.Ecto.Schema.Changeset.user_id_field_changeset/3 update with nil value caused an exception to be raised
	[PowPersistentSession.Plug.Cookie] Now expires the cookie 10 seconds after the last request when authenticating to prevent multiple simultaneous requests deletes the cookie immediately


  
    
  
  Documentation


	Added mailer rate limitation section to production checklist guide
	[Pow.Plug.Session] Added section on session expiration to the docs
	Updated instructions in umbrella project guide to Elixir 1.9
	[Pow.Store.Backend.Base] Updated usage example with Cachex
	Added security practices page


  
    
  
  v1.0.15 (2019-11-20)



  
    
  
  Enhancements


	[Pow.Extension.Base] Extensions are now expected to have a base module with compile-time information whether certain modules are available to prevent unnecessary Code.ensure_compiled?/1 calls:	Added Pow.Extension.Base module
	Added PowEmailConfirmation module
	Added PowInvitation module
	Added PowPersistentSession module
	Added PowResetPassword module


	[PowPersistentSession.Plug.Cookie] Added support for custom metadata:	PowPersistentSession.Plug.Cookie.create/3 now stores a metadata keyword list that can be populated
	PowPersistentSession.Plug.Cookie.create/3 will now, instead of adding :session_fingerprint to the metadata, populate the :session_metadata keyword list with :fingerprint
	PowPersistentSession.Plug.Cookie.authenticate/2 will now populate session metadata with what exists in :session_metadata key for the persistent session metadata
	PowPersistentSession.Plug.Cookie.create/3 now ensures to delete the previous persistent session first, if one is found in cookies


	[Pow.Extension.Config] Added Pow.Extension.Config.extension_modules/2


  
    
  
  Bug fixes


	[Router.Phoenix.Router] Fixed bug where resource routes were not filtered correctly according to the path bindings


  
    
  
  Deprecations


	[Pow.Extension.Config] Deprecated Pow.Extension.Config.discover_modules/2


  
    
  
  v1.0.14 (2019-10-29)



  
    
  
  Changes


	Changed minmum password length to 8 (OWASP/NIST recommendations)
	Pow.Phoenix.Router now only filters routes that has equal number of bindings
	Pow.Phoenix.Routes.user_not_authenticated_path/1 now only puts the :request_path param if the request is using "GET" method
	The stores has been refactored so the command conforms with ETS store. This means that put commands now accept {key, value} record element(s), and keys may be list for easier lookup.	Pow.Store.Backend.Base behaviour now requires to;	Accept Pow.Store.Backend.Base.record/0 values for put/2
	Accept Pow.Store.Backend.Base.key/0 for delete/2 and get/2
	Implement all/2
	Remove keys/1
	Remove put/3


	Pow.Store.Backend.EtsCache now uses :ordered_set instead of :set for efficiency
	Pow.Store.Backend.MnesiaCache now uses :ordered_set instead of :set for efficiency
	Pow.Store.Backend.MnesiaCache will delete all binary key records when initialized
	Pow.Store.Base behaviour now requires to;	Accept erlang term value for keys in all methods
	Implement put/3 instead of put/4
	Implement delete/2 instead of put/3
	Implement get/2 instead of put/3
	Remove keys/2


	Pow.Store.Base.all/3 added
	Pow.Store.Base.put/3 added
	Pow.Store.Base will use binary key rather than key list if all/2 doesn't exist in the backend cache
	Added Pow.Store.CredentialsCache.users/2
	Added Pow.Store.CredentialsCache.sessions/2
	Pow.Store.CredentialsCache now adds a session key rather than appending to a list for the user key to prevent race condition


	Pow.Plug.Session.create/3 now stores a keyword list with metadata for the session rather than just a timestamp
	Pow.Plug.Session.fetch/2 and Pow.Plug.Session.create/3 now assigns :pow_session_metadata in conn.private with the session metadata
	Pow.Plug.Session.create/3 will use the metadata found in conn.private[:pow_session_metadata] if it exists and otherwise add a randomly unique id for :fingerprint
	PowPersistentSession.Plug.Cookie.create/3 will use the value of conn.private[:pow_session_metadata][:fingerprint] if it exists as :session_fingerprint in the persistent session metadata
	PowPersistentSession.Plug.Cookie.authenticate/2 will assign :fingerprint to conn.private[:pow_session_metadata] if it exists in the persistent session metadata
	Pow.Store.CredentialsCache.put/3 will invalidate any other sessions with the same :fingerprint if any is set in session metadata
	PowResetPassword.Phoenix.ResetPasswordController.create/2 when a user doesn't exist will now only return success message if the registration routes has been disabled, otherwise the form with an error message will be returned
	Added PowResetPassword.Phoenix.Messages.user_not_found/1


  
    
  
  Bug fixes


	Fixed bug where Pow.Store.CredentialsCache wasn't used due to how Pow.Store.Base macro worked
	Fixed bug where PowEmailConfirmation.Phoenix.ControllerCallbacks couldn't deliver email


  
    
  
  Deprecations


	Deprecated Pow.Store.Backend.EtsCache.keys/1
	Deprecated Pow.Store.Backend.EtsCache.put/3
	Deprecated Pow.Store.Backend.MnesiaCache.keys/1
	Deprecated Pow.Store.Backend.MnesiaCache.put/3
	Deprecated Pow.Store.Base.keys/2
	Deprecated Pow.Store.Base.put/4
	Deprecated Pow.Store.CredentialsCache.user_session_keys/3
	Deprecated Pow.Store.CredentialsCache.sessions/3


  
    
  
  v1.0.13 (2019-08-25)


	Updated PowEmailConfirmation.Ecto.Schema.changeset/3 so;	when :email is identical to :unconfirmed_email it won't generate new :email_confirmation_token
	when :email is identical to the persisted :email value both :email_confirmation_token and :unconfirmed_email will be set to nil
	when there is no :email value in the params nothing happens


	Updated PowEmailConfirmation.Ecto.Schema.confirm_email_changeset/1 so now :email_confirmation_token is set to nil
	Updated Pow.Ecto.Schema.Changeset.user_id_field_changeset/3 so the e-mail validator now accepts unicode e-mails
	Added PowEmailConfirmation.Ecto.Context.current_email_unconfirmed?/2 and PowEmailConfirmation.Plug.pending_email_change?/1
	Added :email_validator configuration option to Pow.Ecto.Schema.Changeset
	Added Pow.Ecto.Schema.Changeset.validate_email/1
	Fixed bug in PowEmailConfirmation.Phoenix.ControllerCallbacks.send_confirmation_email/2 where the confirmation e-mail wasn't send to the updated e-mail address


  
    
  
  v1.0.12 (2019-08-16)


	Added API integration guide
	Added :reset_password_token_store configuration setting
	To prevent timing attacks, Pow.Ecto.Context.authenticate/2 now verifies password on a blank user struct when no user can be found for the provided user id, but will always return nil. The blank user struct has a nil :password_hash value. The struct will be passed along with a blank password to the verify_password/2 method in the user schema module.
	To prevent timing attacks, when Pow.Ecto.Schema.Changeset.verify_password/3 receives a struct with a nil :password_hash value, it'll hash a blank password, but always return false.
	To prevent timing attacks, the UUID is always generated in PowResetPassword.Plug.create_reset_token/2 whether the user exists or not.
	PowPersistentSession.Plug.Base now accepts :persistent_session_ttl which will pass the TTL to the cache backend and used for the max age of the sesion cookie in PowPersistentSession.Plug.Cookie
	Deprecated :persistent_session_cookie_max_age configuration setting
	Pow.Store.Backend.MnesiaCache can now auto join clusters
	Pow.Store.Backend.MnesiaCache.Unsplit module added for self-healing after network split
	Removed :nodes config option for Pow.Store.Backend.MnesiaCache


  
    
  
  v1.0.11 (2019-06-13)


	Fixed bug in router filters with Phoenix 1.4.7


  
    
  
  v1.0.10 (2019-06-09)


	Prevent browser cache of Pow.Phoenix.SessionController.new/2, Pow.Phoenix.RegistrationController.new/2 and PowInvitation.Phoenix.InvitationController.edit/2 by setting "Cache-Control" header unless it already has been customized
	All links in docs generated with mix docs and on hexdocs.pm now works
	Generated docs now uses lower case file name except for README, CONTRIBUTING and CHANGELOG
	Removed duplicate call for Pow.Plug.Session.delete/2 in Pow.Plug.Sesssion.create/3


  
    
  
  v1.0.9 (2019-06-04)



  
    
  
  Changes


	Pow.Phoenix.Router will now only add specific routes if there is no matching route already defined
	Added Pow.Plug.get_plug/1 and instead of :mod, :plug is used in config
	Pow.Ecto.Context.authenticate/2 now returns nil if user id or password is nil


  
    
  
  Bug fixes


	Fixed bug with exception raised in Pow.Ecto.Schema.normalize_user_id_field_value/1 when calling Pow.Ecto.Context.get_by/2 with a non binary user id
	Fixed bug with exception raised in Pow.Ecto.Schema.normalize_user_id_field_value/1 when calling Pow.Ecto.Context.authenticate/2 with a non binary user id


  
    
  
  Deprecations


	Deprecated Pow.Plug.get_mod/1
	Removed call to Pow.Ecto.Context.repo/1


  
    
  
  v1.0.8 (2019-05-24)



  
    
  
  Changes


	Added support for layout in mails with Pow.Phoenix.Mailer.Mail by setting conn.private[:pow_mailer_layout] same way as the Phoenix layout with conn.private[:phoenix_layout]
	Added :prefix repo opts support to use in multitenant apps
	Removed @changeset.data.__struct__.pow_user_id_field() in template in favor of using Pow.Ecto.Schema.user_id_field/1


  
    
  
  Bug fixes


	Fixed bug in Pow.Ecto.Schema.Changeset.current_password_changeset/3 where an exception would be thrown if the virtual :current_password field of the user struct was set and either the :current_password change was blank or identical


  
    
  
  Deprecations


	Deprecated Mix.Pow.Ecto.Migration.create_migration_files/3 and moved it to Mix.Pow.Ecto.Migration.create_migration_file/3
	Deprecated Pow.Ecto.Context.repo/1 and moved it to Pow.Config.repo!/1
	Deprecated Pow.Ecto.Context.user_schema_mod/1 and moved it to Pow.Config.user!/1


  
    
  
  v1.0.7 (2019-05-01)


	Fixed bug with Phoenix 1.4.4 scoped routes


  
    
  
  v1.0.6 (2019-04-19)


	Fixed bug where custom layout setting raised exception in Pow.Phoenix.ViewHelpers.layout/1
	Prevent users from changing their email to one already taken when the PowEmailConfirmation extension has been enabled


  
    
  
  v1.0.5 (2019-04-09)


	Added extension_messages/1 to extension controllers and callbacks
	Improved feedback for when no templates are generated for an extension with mix pow.extension.phoenix.gen.templates and mix pow.extension.phoenix.mailer.gen.templates tasks
	Error flash is no longer overridden in Pow.Phoenix.PlugErrorHandler if the error message is nil
	Fixed bug in the migration generator where references/2 wasn't called with options
	Support any :plug version below 2.0.0
	Deprecated Pow.Extension.Ecto.Context.Base


  
    
  
  v1.0.4 (2019-03-13)


	Added PowInvitation to the mix pow.extension.phoenix.gen.templates and mix pow.extension.phoenix.mailer.gen.templates tasks
	Fixed issue in umbrella projects where extensions wasn't found in environment configuration
	Fixed so :namespace environment config can be used as web app module name
	Shell instructions will only be printed if the configuration is missing
	Now requires that :ecto or :phoenix are included in the dependency list for the app to run respective mix tasks
	Deprecated Mix.Pow.context_app/0
	Deprecated Mix.Pow.ensure_dep!/3
	Deprecated Mix.Pow.context_base/1


  
    
  
  v1.0.3 (2019-03-09)



  
    
  
  Changes


	Added PowInvitation extension
	Added support in Pow.Ecto.Schema for Ecto associations fields
	Added support for adding custom methods with Pow.Extension.Ecto.Schema through __using__/1 macro in extension ecto schema module
	Help information raised with invalid schema arguments for pow.install, pow.ecto.install, pow.ecto.gen.migration, and pow.ecto.gen.schema mix tasks
	PowEmailConfirmation now redirects unconfirmed users to after_registration_path/1 or after_sign_in_path/1 rather than pow_session_path(conn, :new)


  
    
  
  Bug fixes


	Correct shell instructions for mix pow.install task with custom schema
	Fixed bug in Pow.Extension.Phoenix.Router.Base and Pow.Extension.Phoenix.Messages where the full extension name wasn't used to namespace methods


  
    
  
  Deprecations


	Deprecated Pow.Extension.Config.underscore_extension/1
	Deprecated PowResetPassword.Ecto.Context.password_changeset/2
	Deprecated Pow.Ecto.Schema.filter_new_fields/2
	Deprecated :messages_backend_fallback setting for extension controllers
	Removed deprecated macro router_helpers/1 in Pow.Phoenix.Controller


  
    
  
  v1.0.2 (2019-02-28)


	Added flash error message about e-mail confirmation for when user changes e-mail with PowEmailConfirmation enabled
	Added new_password_changeset/3 and confirm_password_changeset/3 to Pow.Ecto.Schema.Changeset
	Redis cache store backend guide
	Correct shell instructions for mix pow.phoenix.gen.templates task
	Only load environment config in Pow.Config.get/3 when no key is set in the provided config
	Fixed issue in Pow.Store.Backend.MnesiaCache.keys/1 and Pow.Store.Backend.EtsCache.keys/1 so they now return keys without namespace
	Pow.Store.Backend.MnesiaCache.put/3 now raises an error if TTL is not provided


  
    
  
  Breaking changes


	PowResetPassword.Plug.reset_password_token/1 has been removed


  
    
  
  v1.0.1 (2019-01-27)


	pow.extension.ecto.gen.migrations mix task will output warning when a migration file won't be generated for any particular extension
	Leading and trailing whitespace is removed from the user id field value (in addition to forced lower case)
	An exception is raised when pow_routes/0 or pow_extension_routes/0 are used inside router scopes with aliases
	Mail view templates assigns now has [user: user, conn: conn] along with the template specific assigns
	Mail view subject methods now gets the same assigns passed as mail view template assigns instead of only [conn: conn]
	Added pow_registration_routes/0, pow_session_routes/0 and pow_scope/1 macros to the router module
	Added guide on how to disable registration


  
    
  
  v1.0.0 (2018-11-18)


	Phoenix 1.4 support
	Ecto 3.0 support



  

    
PowEmailConfirmation
    

This extension will send an e-mail confirmation link when the user registers, and when the user changes their e-mail. It requires that the user schema has an :email field.
Users won't be signed in when they register, and can't sign in until the e-mail has been confirmed. The confirmation e-mail will be sent every time the sign in fails. The user will be redirected to after_registration_path/1 and after_sign_in_path/1 accordingly.
To prevent user enumeration, the user will see the same confirmation required message if the account couldn't be created due to unique constraint error on :email. No e-mail will be sent. If pow_prevent_user_enumeration: false is set in conn.private the form with error will be shown instead.
When users updates their e-mail, it won't be changed until the user has confirmed the new e-mail by clicking the e-mail confirmation link. The confirmation will fail if the :email is already in use for another account. If PowInvitation is used then the same logic applies when a user accepts an invitation changing their e-mail address in the process.

  
    
  
  Installation


Follow the instructions for extensions in README.md, and set PowEmailConfirmation in the :extensions list.

  
    
  
  Configuration


Add the following section to your WEB_PATH/controllers/pow/registration_html/edit.html.heex template (you may need to generate the templates first) after the pow_user_id_field field:
<p :if={@changeset.data.unconfirmed_email} class="phx-no-feedback:hidden mt-3 flex gap-3 text-sm leading-6 text-rose-600">
  <.icon name="hero-exclamation-circle-mini" class="mt-0.5 w-5 h-5 flex-none" />
  Click the link in the confirmation email to change your email to <span class="font-semibold text-brand hover:underline">@changeset.data.unconfirmed_email)</span>.
</p>

  
    
  
  Prevent persistent session sign in


To prevent that PowPersistentSession creates a new persistent session when the email hasn't been confirmed, PowEmailConfirmation should be placed first in the extensions list. It'll halt the connection.

  
    
  
  Test and seed


If you want your user to be automatically confirmed in test and seed, you just have to call: PowEmailConfirmation.Ecto.Context.confirm_email(user, %{}, otp_app: :my_app)
You can also update or insert the row directly and set email_confirmed_at: DateTime.utc_now().

  
    
  
  Note on PowInvitation


When a user is invited with PowInvitation, the email will only be required confirmed if the invited user decides to change their email when accepting the invitation.


  

    
PowInvitation
    

This extension will set up a basic invitation system where users can invite other users to join. If :email field exists on the user struct an e-mail is sent out with an invitation link. Otherwise a page with the invitation link is shown.
Invited users are persisted in the database without a password. Only the user id will be validated when the user is invited, but changeset/2 on your user schema will be used for when the user accepts the invitation.
To prevent user enumeration, the invited user will, to the inviter, appear as succesfully invited even if the user couldn't be created due to unique constraint error on :email. No e-mail will be sent out. If pow_prevent_user_enumeration: false is set in conn.private the form with error will be shown instead.
An invited user can change their e-mail when accepting the invitation. To prevent user enumeration PowEmailConfirmation extension can be enabled.

  
    
  
  Installation


Follow the instructions for extensions in README.md, and set PowInvitation in the :extensions list.

  
    
  
  Configuration


There are numerous ways you can modify the invitation flow. Here's a few common setups to get your started.

  
    
  
  Parent organization or team


If your users belongs to a parent organization or team, you can set up the invite_changeset/3 to carry over the id for invitations:
defmodule MyApp.Users.User do
  # ...

  def invite_changeset(user_or_changeset, invited_by, attrs) do
    user_or_changeset
    |> pow_invite_changeset(invited_by, attrs)
    |> changeset_organization(invited_by)
  end

  defp changeset_organization(changeset, invited_by) do
    Ecto.Changeset.change(changeset, organization_id: invited_by.organization_id)
  end
end

  
    
  
  Limit invitation based on role


If you have different roles (e.g. admin and user), you can limit the type of user who can invite others by using a plug and override the routes in your router.ex:
defmodule MyAppWeb.Router do
  use MyAppWeb, :router
  # ...

  pipeline :admin_role do
    plug MyAppWeb.EnsureRolePlug, :admin
  end

  scope "/", PowInvitation.Phoenix, as: "pow_invitation" do
    pipe_through [:browser, :protected, :admin_role]

    resources "/invitations", InvitationController, only: [:new, :create, :show]
  end

  # ... you would want `pow_extension_routes/0` with the default routes to be after this
end
The routes will override the default ones in pow_extension_routes/0.

  
    
  
  Limit registration


If you wish to restrict signup to only invites, you can modify router.ex to exclude registration by using pow_session_routes/0 in place of pow_routes/0:
defmodule MyAppWeb.Router do
  use MyAppWeb, :router
  # ...

  scope "/" do
    pipe_through :browser

    pow_session_routes()
    pow_extension_routes()
  end

  # ...
end
You need to  use custom template for the session controller, since by default it'll still have the link to registration. Read more in the Disable registration guide.

  
    
  
  Expire invited users


Invited users will have a token set in the :invitation_token field and :invitation_accepted_at field set to nil. If you want to expire the invitation link you can run a background task to delete these users a certain time after :inserted_at.

  
    
  
  Note on PowEmailConfirmation


PowEmailConfirmation will only require email confirmation if the invited user changes their email when accepting their invitation.


  

    
PowPersistentSession
    

This extension can reissue sessions by setting a cookie with a token that may be used exactly once to issue the session. The cookie and token will expire after 30 days. Once the session has been issued, a new cookie and token will be set that expires after another 30 days.

  
    
  
  Installation


Follow the instructions for extensions in README.md, and set PowPersistentSession in the :extensions list.
Add the following plug after the pow session plug in your WEB_PATH/endpoint.ex:
defmodule MyAppWeb.Endpoint do
  # ...

  plug Pow.Plug.Session, otp_app: :my_app
  plug PowPersistentSession.Plug.Cookie
  #...
end

  
    
  
  Configuration


By default, the persistent session is automatically used if the extension has been enabled. If you wish to let the user manage this, you should add the following checkbox to the form in WEB_PATH/controllers/pow/session_html/new.html.heex (you may need to generate the templates first):
<.input field={f[:persistent_session]} type="checkbox" label="Keep me logged in" />


  

    
PowResetPassword
    

This extension will allow users to reset the password by sending an e-mail with a reset password link. It requires that the user schema has an :email field.
To prevent user enumeration attacks, the generic PowResetPassword.Phoenix.Messages.maybe_email_has_been_sent/1 message is always shown when requesting password reset. If pow_prevent_user_enumeration: false is set in conn.private the form will be shown instead with the PowResetPassword.Phoenix.Messages.user_not_found/1 message.

  
    
  
  Installation


Follow the instructions for extensions in README.md, and set PowResetPassword in the :extensions list.

  
    
  
  Configuration


Add the following link to your WEB_PATH/controllers/pow/session_html/new.html.heex template (you may need to generate the templates first):
<.link navigate={~p"/reset-password/new"} class="text-sm font-semibold">Forgot your password?<./link>


  

    
Why Pow
    

Why should you use Pow instead of any other user management system? Let's take a look at what Pow does differently.

  
    
  
  Functional configuration


No global configuration!
Unlike most alternative user management libraries, Pow is built to be a functional component in your app. You can have numerous separate Pow setups in the same app, e.g., a super admin backend and a regular user login.
Configuration can be passed to nearly all functions in runtime or compile time.

  
    
  
  Modular


Pow has an explicit API for its Plug, Ecto and Phoenix modules. Any part can be removed entirely, or modified. Each of the main category modules has a clear separation of responsibilities that makes it easy to micro adjust the flow.

  
    
  
  Low dependency requirement


Nearly everything is baked in. You won't get into annoying dependency conflicts!

  
    
  
  Extendable


Pow starts out simple: Session management and user registration.
But with Pow you've access to multiple extensions that covers most needs for user management: E-mail confirmation, reset password and long-term session (remember me). There's also a multi-provider library that can be used, e.g. with Twitter and Github.

  
    
  
  Developer friendly


With a clear API in mind, all instructions are written so it's easy to understand the basics of Pow. As you get your first app with Pow up and running, you'll understand the underlying mechanisms. Pow has extensive documentation in all its modules that will clearly show you how Pow works.
Explicit rather than implicit.

  
    
  
  Secure


The basis for security in Pow is industry practices and recommendations from, among others, NIST and OWASP. You can read more in the README.md for specifics.

  
    
  
  Production ready


There are several production apps successfully running Pow. From migration from Coherence to highly customized Pow setups.


  

    
Production checklist
    

Before you deploy your Pow enabled app to production, you should take a look at the following checklist to ensure that your app is ready.
The list is not exhaustive, and you should take any appropriate additional steps for your particular setup.

  
    
  
  REQUIRED: Use a persistent cache store


By default the Pow.Store.Backend.EtsCache will be used as the cache backend. In production, this would mean that all session data will be lost between deploys or server restarts. Furthermore, in clusters, the sessions will not be shared between nodes.
You should use a persistent (and possibly distributed) cache store like the Pow.Store.Backend.MnesiaCache.
To enable the Mnesia cache you should add it to your application.ex supervisor:
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      MyApp.Repo,
      MyAppWeb.Endpoint,
      Pow.Store.Backend.MnesiaCache
      # # Or in a distributed system:
      # {Pow.Store.Backend.MnesiaCache, extra_db_nodes: {Node, :list, []}},
      # Pow.Store.Backend.MnesiaCache.Unsplit # Recover from netsplit
    ]

    opts = [strategy: :one_for_one, name: MyAppWeb.Supervisor]
    Supervisor.start_link(children, opts)
  end

  # ...
end
Update the config with cache_store_backend: Pow.Store.Backend.MnesiaCache.
Mnesia will store the database files in the directory ./Mnesia.NODE in the current working directory where NODE is the node name. By default, this is ./Mnesia.nonode@nohost. You may wish to change the location to a shared directory so you can roll deploys:
config :mnesia, :dir, ~c"/path/to/dir"
:mnesia should be added to :extra_applications in mix.exs for it to be included in releases.

  
    
  
  OPTIONAL: Validate that strong passwords are used


NIST 800-63b recommends that you reject passwords that are commonly-used, expected, or compromised. The guidelines explicitly mention the following methods to ensure strong passwords are used:
	Passwords obtained from previous breach corpuses.
	Dictionary words.
	Repetitive or sequential characters (e.g. ‘aaaaaa’, ‘1234abcd’).
	Context-specific words, such as the name of the service, the username, and derivatives thereof.


You can read how to handle password breach lookup and other NIST based validation rules on the powauth.com website.

  
    
  
  OPTIONAL: Use an appropriate password hash method


By default PBKDF2-SHA512 with 100,000 iterations is used for password hashing. This is what's recommended by NIST 800-63b. If you are allowed to use other password hashing algorithms, then Argon2id is considered a safer option.
You can easily change the password hashing method in Pow. Here's how you can use comeonin with Argon2:
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema,
    password_hash_verify: {&Argon2.hash_pwd_salt/1,
                           &Argon2.verify_pass/2}

  # ...
end

  
    
  
  OPTIONAL: Rate limit authentication attempts


You should rate limit authentication attempts according to NIST 800-63b. Pow doesn't include a rate limiter since this is better dealt with at the proxy or gateway side rather than the application side. The minimum requirement would be to rate limit to a maximum of 100 failed authentication attempts per IP.
You may also wish to lock accounts that has had too many failed authentication attempts, or require a CAPTCHA to be solved before allowing new attempts.

  
    
  
  OPTIONAL: Rate limit e-mail delivery


There are no rate limits for any e-mails sent out with Pow, including PowEmailConfirmation, PowInvitation and PowResetPassword extensions. If you use a transactional e-mail service you have to make careful considerations to prevent resource usage attacks.
Rate limitation should either be handled at the service, or you may be able to set up rate limitation in the Pow mailer. For the latter, here's a simple example using Hammer:
defmodule MyAppWeb.Pow.Mailer do
  use Pow.Phoenix.Mailer

  # ....

  require Logger

  @impl true
  def process(email) do
    case check_rate(email) do
      {:allow, _count} -> deliver(email)
      {:deny, _count}  -> Logger.warning("Mailer backend failed due to rate limitation: #{inspect(email)}")
    end
  end

  defp check_rate(%{to: email}) do
    Hammer.check_rate_inc("email:#{email}", :timer.minutes(1), 2, 1)
  end
end
In the above, the e-mail delivery will be limited to two e-mails per minute for a single recipient, but you can use different criteria, e.g. limit for e-mails that have the same recipient and subject. It's strongly recommended to add tests where appropriate to ensure abuse is not possible.


  

    
Security practices
    

Some of the below is based on OWASP or NIST SP800-63b recommendations.

  
    
  
  User ID


	The user_id_field value is always treated as case insensitive
	If the user_id_field is :email, it'll be validated based on RFC 5322 (sections 3.2.3 and 3.4.1) and RFC 5321 with unicode characters permitted in local and domain part


  
    
  
  Password


	The :password has a minimum length of 8 characters
	The :password has a maximum length of 4096 bytes to prevent DOS attacks against Pbkdf2
	The :password_hash is generated with PBKDF2-SHA512 with 100,000 iterations


  
    
  
  Session management


	The session value contains a UUID token that is used to pull credentials through a GenServer
	The credentials are stored in a key-value cache with TTL of 30 minutes
	The credentials and session are renewed after 15 minutes if any activity is detected
	The credentials and session are renewed when user updates


  
    
  
  Timing attacks


	If a user couldn't be found or the :password_hash is nil a blank password is used
	A UUID is always generated during reset password flow
	Tokens are signed for public consumption and verified before lookup:	Session ID in Pow.Plug.Session
	Persistent session token in PowPersistentSession.Plug.Cookie
	Reset password token in PowResetPassword.Plug
	E-mail confirmation token in PowEmailConfirmation.Plug
	Invitation token in PowInvitation.Plug




  
    
  
  User enumeration attacks


	If authentication fails, a generic The provided login details did not work. Please verify your credentials, and try again. message is returned
	When password reset is requested with PowResetPassword for an e-mail that doesn't exist, the generic If an account for the provided email exists, an email with reset instructions will be send to you. Please check your inbox. message is returned
	When attempting to invite a user with PowInvitation using an already taken e-mail, the success message An e-mail with invitation link has been sent. is returned

Enabling PowEmailConfirmation extension will add additional protection:
	User is redirected with message to confirm their e-mail when they attempt to create a user with already taken e-mail
	Updating e-mail requires the user to confirm the e-mail address by clicking a link send to them

You can disable the protection by setting pow_prevent_user_enumeration: false in conn.private.

  
    
  
  Browser cache


	The sign in, registration and invitation acceptance page won't be cached by the browser



  

    
Configuring mailer
    

You are able to configure a mailer for Pow and set it up with the library of 
your choice.
This guide shows how to setup Pow with
	Swoosh
	Bamboo

You must first setup and configure either of these libraries before you can
integrate them with Pow.

  
    
  
  Swoosh mailer


Set up your WEB_PATH/mails/pow_mailer.ex file like so:
defmodule MyAppWeb.Pow.Mailer do
  use Pow.Phoenix.Mailer
  use Swoosh.Mailer, otp_app: :my_app

  import Swoosh.Email
  
  require Logger

  @impl true
  def cast(%{user: user, subject: subject, text: text, html: html}) do
    %Swoosh.Email{}
    |> to({"", user.email})
    |> from({"My App", "myapp@example.com"})
    |> subject(subject)
    |> html_body(html)
    |> text_body(text)
  end

  @impl true
  def process(email) do
    # An asynchronous process should be used here to prevent enumeration
    # attacks. Synchronous e-mail delivery can reveal whether a user already
    # exists in the system or not.

    Task.start(fn ->
      email
      |> deliver()
      |> log_warnings()
    end)

    :ok
  end

  defp log_warnings({:error, reason}) do
    Logger.warning("Mailer backend failed with: #{inspect(reason)}")
  end

  defp log_warnings({:ok, response}), do: {:ok, response}
end
Remember to add mailer_backend: MyAppWeb.Pow.Mailer to the Pow configuration, and set the Swoosh configuration:
# config/config.exs
config :my_app, :pow,
  user: MyApp.Users.User,
  repo: MyApp.Repo,
  mailer_backend: MyAppWeb.Pow.Mailer

# config/prod.exs
config :my_app, MyAppWeb.Pow.Mailer,
  adapter: Swoosh.Adapters.Sendgrid,
  api_key: "SG.x.x"

  
    
  
  Bamboo mailer


Set up your WEB_PATH/mails/pow_mailer.ex file like so:
defmodule MyAppWeb.Pow.Mailer do
  use Pow.Phoenix.Mailer
  use Bamboo.Mailer, otp_app: :my_app

  import Bamboo.Email

  @impl true
  def cast(%{user: user, subject: subject, text: text, html: html}) do
    new_email(
      to: user.email,
      from: "myapp@example.com",
      subject: subject,
      html_body: html,
      text_body: text
    )
  end

  @impl true
  def process(email) do
    # An asynchronous process should be used here to prevent enumeration
    # attacks. Synchronous e-mail delivery can reveal whether a user already
    # exists in the system or not.

    deliver_later(email)
  end
end
Remember to add mailer_backend: MyAppWeb.Pow.Mailer to the Pow configuration, and set the Bamboo configuration:
# config/config.exs
config :my_app, :pow,
  user: MyApp.Users.User,
  repo: MyApp.Repo,
  mailer_backend: MyAppWeb.Pow.Mailer

# config/prod.exs
config :my_app, MyAppWeb.Pow.Mailer,
  adapter: Bamboo.MandrillAdapter, # Specify your preferred adapter
  api_key: "my_api_key" # Specify adapter-specific configuration


  

    
How to add user roles
    

Adding user roles is very simple, and you won't need an extension for this. This can be done in several ways, but we'll work with the most straight forward setup.

  
    
  
  Update your schema


Add a role column to your user schema. In our example, we'll set the default role to user and add a changeset_role/2 function that ensures the role can only be user or admin.
# lib/my_app/users/user.ex
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema

  schema "users" do
    field :role, :string, default: "user"

    pow_user_fields()

    timestamps()
  end

  @spec changeset_role(Ecto.Schema.t() | Ecto.Changeset.t(), map()) :: Ecto.Changeset.t()
  def changeset_role(user_or_changeset, attrs) do
    user_or_changeset
    |> Ecto.Changeset.cast(attrs, [:role])
    |> Ecto.Changeset.validate_inclusion(:role, ~w(user admin))
  end
end
To keep your app secure you shouldn't allow any direct calls to changeset_role/2 with params provided by the user. Instead you should set up functions in your users context module to either create an admin user or update the role of an existing user:
# lib/my_app/users.ex
defmodule MyApp.Users do
  alias MyApp.{Repo, Users.User}

  @type t :: %User{}

  @spec create_admin(map()) :: {:ok, t()} | {:error, Ecto.Changeset.t()}
  def create_admin(params) do
    %User{}
    |> User.changeset(params)
    |> User.changeset_role(%{role: "admin"})
    |> Repo.insert()
  end

  @spec set_admin_role(t()) :: {:ok, t()} | {:error, Ecto.Changeset.t()}
  def set_admin_role(user) do
    user
    |> User.changeset_role(%{role: "admin"})
    |> Repo.update()
  end
end
Now you can safely call either MyApp.Users.create_admin/1 or MyApp.Users.set_admin_role/1 from your controllers.

  
    
  
  Add role plug


This is all the work you'll need to control access:
# lib/my_app_web/ensure_role_plug.ex
defmodule MyAppWeb.EnsureRolePlug do
  @moduledoc """
  This plug ensures that a user has a particular role.

  ## Example

      plug MyAppWeb.EnsureRolePlug, [:user, :admin]

      plug MyAppWeb.EnsureRolePlug, :admin

      plug MyAppWeb.EnsureRolePlug, ~w(user admin)a
  """
  import Plug.Conn, only: [halt: 1]

  use MyAppWeb, :verified_routes

  alias Phoenix.Controller
  alias Plug.Conn
  alias Pow.Plug

  @doc false
  @spec init(any()) :: any()
  def init(config), do: config

  @doc false
  @spec call(Conn.t(), atom() | binary() | [atom()] | [binary()]) :: Conn.t()
  def call(conn, roles) do
    conn
    |> Plug.current_user()
    |> has_role?(roles)
    |> maybe_halt(conn)
  end

  defp has_role?(nil, _roles), do: false
  defp has_role?(user, roles) when is_list(roles), do: Enum.any?(roles, &has_role?(user, &1))
  defp has_role?(user, role) when is_atom(role), do: has_role?(user, Atom.to_string(role))
  defp has_role?(%{role: role}, role), do: true
  defp has_role?(_user, _role), do: false

  defp maybe_halt(true, conn), do: conn
  defp maybe_halt(_any, conn) do
    conn
    |> Controller.put_flash(:error, "Unauthorized access")
    |> Controller.redirect(to: ~p"/")
    |> halt()
  end
end
Now you can add plug MyAppWeb.EnsureRolePlug, :admin to your pipeline in router.ex:
# lib/my_app_web/router.ex
defmodule MyAppWeb.Router do
  use MyAppWeb, :router
  use Pow.Phoenix.Router

  # ...

  pipeline :admin do
    plug MyAppWeb.EnsureRolePlug, :admin
  end

  scope "/admin", MyAppWeb do
    pipe_through [:browser, :admin]

    # ...
  end

  # ...
end
Or you can add it to your controller(s):
# lib/my_app_web/controllers/some_controller.ex
defmodule MyAppWeb.SomeController do
  use MyAppWeb, :controller

  # ...

  plug MyAppWeb.EnsureRolePlug, :admin

  # ...
end

  
    
  
  Using role permissions with layout


You may wish to render certain sections in your layout for certain roles. First let's add a helper function to the users context:
# lib/my_app/users.ex
defmodule MyApp.Users do
  alias MyApp.{Repo, Users.User}

  # ...

  @spec is_admin?(t()) :: boolean()
  def is_admin?(%{role: "admin"}), do: true
  def is_admin?(_any), do: false
end
Now we can use that helper to conditionally render a section in our templates:
<%= if MyApp.Users.is_admin?(@current_user) do %>
  <div>
    <p>You have admin access</p>
  </div>
<% end %>

  
    
  
  Test modules


# test/my_app/users/user_test.exs
defmodule MyApp.Users.UserTest do
  use MyApp.DataCase

  alias MyApp.Users.User

  test "changeset/2 sets default role" do
    user =
      %User{}
      |> User.changeset(%{})
      |> Ecto.Changeset.apply_changes()

    assert user.role == "user"
  end

  test "changeset_role/2" do
    changeset = User.changeset_role(%User{}, %{role: "invalid"})
    assert changeset.errors[:role] == {"is invalid", [validation: :inclusion, enum: ["user", "admin"]]}

    changeset = User.changeset_role(%User{}, %{role: "admin"})
    refute changeset.errors[:role]
  end
end
# test/my_app/users_test.exs
defmodule MyApp.UsersTest do
  use MyApp.DataCase

  alias MyApp.{Repo, Users, Users.User}

  @valid_params %{email: "test@example.com", password: "secret1234", password_confirmation: "secret1234"}

  test "create_admin/2" do
    assert {:ok, user} = Users.create_admin(@valid_params)
    assert user.role == "admin"
  end

  test "set_admin_role/1" do
    assert {:ok, user} = Repo.insert(User.changeset(%User{}, @valid_params))
    assert user.role == "user"

    assert {:ok, user} = Users.set_admin_role(user)
    assert user.role == "admin"
  end

  # Uncomment if you added this function to your users context
  # test "is_admin?/1" do
  #   refute Users.is_admin?(nil)
  #
  #   assert {:ok, user} = Repo.insert(User.changeset(%User{}, @valid_params))
  #   refute Users.is_admin?(user)
  #
  #   assert {:ok, admin} = Users.create_admin(%{@valid_params | email: "test2@example.com"})
  #   assert Users.is_admin?(admin)
  # end
end
# test/my_app_web/ensure_role_plug_test.exs
defmodule MyAppWeb.EnsureRolePlugTest do
  use MyAppWeb.ConnCase

  alias MyAppWeb.EnsureRolePlug

  @opts ~w(admin)a
  @user %{id: 1, role: "user"}
  @admin %{id: 2, role: "admin"}

  setup do
    conn =
      build_conn()
      |> Plug.Conn.put_private(:plug_session, %{})
      |> Plug.Conn.put_private(:plug_session_fetch, :done)
      |> Pow.Plug.put_config(otp_app: :my_app)
      |> fetch_flash()

    {:ok, conn: conn}
  end

  test "call/2 with no user", %{conn: conn} do
    opts = EnsureRolePlug.init(@opts)
    conn = EnsureRolePlug.call(conn, opts)

    assert conn.halted
    assert redirected_to(conn) == ~p"/"
  end

  test "call/2 with non-admin user", %{conn: conn} do
    opts = EnsureRolePlug.init(@opts)
    conn =
      conn
      |> Pow.Plug.assign_current_user(@user, otp_app: :my_app)
      |> EnsureRolePlug.call(opts)

    assert conn.halted
    assert redirected_to(conn) == ~p"/"
  end

  test "call/2 with non-admin user and multiple roles", %{conn: conn} do
    opts = EnsureRolePlug.init(~w(user admin)a)
    conn =
      conn
      |> Pow.Plug.assign_current_user(@user, otp_app: :my_app)
      |> EnsureRolePlug.call(opts)

    refute conn.halted
  end

  test "call/2 with admin user", %{conn: conn} do
    opts = EnsureRolePlug.init(@opts)
    conn =
      conn
      |> Pow.Plug.assign_current_user(@admin, otp_app: :my_app)
      |> EnsureRolePlug.call(opts)

    refute conn.halted
  end
end


  

    
How to lock users
    

Locking users is trivial, and you won't need an extension for this. It can be done in several ways, but we'll work with the most straight forward setup.

  
    
  
  Update your schema


Add a locked_at column to your user schema, and a lock_changeset/1 function to lock the account:
# lib/my_app/users/user.ex
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema
  
  alias Ecto.{Changeset, Schema}

  schema "users" do
    field :locked_at, :utc_datetime

    pow_user_fields()

    timestamps()
  end

  @spec lock_changeset(Schema.t() | Changeset.t()) :: Changeset.t()
  def lock_changeset(user_or_changeset) do
    changeset = Changeset.change(user_or_changeset)
    locked_at = DateTime.truncate(DateTime.utc_now(), :second)

    case Changeset.get_field(changeset, :locked_at) do
      nil  -> Changeset.change(changeset, locked_at: locked_at)
      _any -> Changeset.add_error(changeset, :locked_at, "already set")
    end
  end
end
Add a lock action to your user context module:
# lib/my_app/users.ex
defmodule MyApp.Users do
  alias MyApp.{Repo, Users.User}

  @spec lock(map()) :: {:ok, map()} | {:error, map()}
  def lock(user) do
    user
    |> User.lock_changeset()
    |> Repo.update()
  end
end

  
    
  
  Set up controller


Create or modify you user management controller so you (likely the admin) can lock the account:
# lib/my_app_web/controllers/admin/user_controller.ex
defmodule MyAppWeb.Admin.UserController do
  use MyAppWeb, :controller

  alias Plug.Conn
  alias Pow.{Plug, Operations}
  alias MyApp.Users

  plug :load_user when action in [:lock]

  # ...

  @spec lock(Conn.t(), map()) :: Conn.t()
  def lock(%{assigns: %{user: user}} = conn, _params) do
    case Users.lock(user) do
      {:ok, _user} ->
        conn
        |> put_flash(:info, "User has been locked.")
        |> redirect(to: "/")

      {:error, _changeset} ->
        conn
        |> put_flash(:error, "User couldn't be locked.")
        |> redirect(to: "/")
    end
  end

  defp load_user(%{params: %{"id" => user_id}} = conn, _opts) do
    config = Plug.fetch_config(conn)

    case Operations.get_by([id: user_id], config) do
      nil ->
        conn
        |> put_flash(:error, "User doesn't exist")
        |> redirect(to: "/")

      user ->
        assign(conn, :user, user)
    end
  end
end
Remember to add this route to your router.ex file:
# lib/my_app_web/router.ex
defmodule MyAppWeb.Router do
  use MyAppWeb, :router
  use Pow.Phoenix.Router

  # ...

  scope "/admin", MyAppWeb.Admin, as: :admin do
    pipe_through :browser
    # Usually you would lock this area with a plug:
    # pipe_through [:browser, :require_admin_user]

    post "/users/:id/lock", UserController, :lock
  end

  # ...
end

  
    
  
  Prevent sign in for locked users


This is all you need to ensure locked users can't sign in:
# lib/my_app_web/ensure_user_not_locked_plug.ex
defmodule MyAppWeb.EnsureUserNotLockedPlug do
  @moduledoc """
  This plug ensures that a user isn't locked.

  ## Example

      plug MyAppWeb.EnsureUserNotLockedPlug
  """
  import Plug.Conn, only: [halt: 1]

  use MyAppWeb, :verified_routes
  alias Phoenix.Controller
  alias Plug.Conn
  alias Pow.Plug

  @doc false
  @spec init(any()) :: any()
  def init(opts), do: opts

  @doc false
  @spec call(Conn.t(), any()) :: Conn.t()
  def call(conn, _opts) do
    conn
    |> Plug.current_user()
    |> locked?()
    |> maybe_halt(conn)
  end

  defp locked?(%{locked_at: locked_at}) when not is_nil(locked_at), do: true
  defp locked?(_user), do: false

  defp maybe_halt(true, conn) do
    conn
    |> Plug.delete()
    |> Controller.put_flash(:error, "Sorry, your account is locked.")
    |> Controller.redirect(to: ~p"/session/new")
    |> halt()
  end
  defp maybe_halt(_any, conn), do: conn
end
Add plug MyAppWeb.EnsureUserNotLockedPlug to your endpoint or pipeline, and presto!

  
    
  
  Optional: PowResetPassword


The above will prevent any locked users access, but it doesn't prevent them from using features that doesn't require authentication such as resetting their password. Be advised that this is a entirely optional step as this only affects UX.
While there are many different ways of handling this, the most explicit one is to simply override the logic entirely with a custom controller:
# lib/my_app_web/controllers/reset_password_controller.ex
defmodule MyAppWeb.ResetPasswordController do
  use MyAppWeb, :controller

  alias PowResetPassword.{Phoenix.ResetPasswordController, Plug, Store.ResetTokenCache}

  def create(conn, params) do
    conn
    |> ResetPasswordController.process_create(params)
    |> maybe_halt()
    |> ResetPasswordController.respond_create()
  end

  defp maybe_halt({:ok, %{token: token, user: %{locked_at: locked_at}}, conn}) when not is_nil(locked_at) do
    user = Plug.change_user(conn)

    expire_token(conn, token)

    {:error, %{user | action: :update}, conn}
  end
  defp maybe_halt(response), do: response

  defp expire_token(conn, token) do
    config = Pow.Plug.fetch_config(conn)

    case Pow.Plug.verify_token(conn, Atom.to_string(PowResetPassword.Plug), token, config) do
      {:ok, token} ->
        backend = Pow.Config.get(config, :cache_store_backend, Pow.Store.Backend.EtsCache)

        ResetTokenCache.delete([backend: backend], token)

      :error ->
        :ok
    end
  end
end
To make the code simpler for us we're leveraging the functions from PowResetPassword.Phoenix.ResetPasswordController here.
Now all we got to do is to catch the route before the pow_extension_routes/0 call:
# lib/my_app_web/router.ex
defmodule MyAppWeb.Router do
  use MyAppWeb, :router
  use Pow.Phoenix.Router
  use Pow.Extension.Phoenix.Router,
    extensions: [PowResetPassword]

  # ...

  scope "/", MyAppWeb do
    pipe_through :browser

    post "/reset-password", ResetPasswordController, :create
  end

  scope "/" do
    pow_routes()
    pow_extension_routes()
  end

  # ...
end

  
    
  
  Test modules


# test/my_app/users_test.exs
defmodule MyApp.UsersTest do
  use MyApp.DataCase

  alias MyApp.{Repo, Users, Users.User}

  @valid_params %{email: "test@example.com", password: "secret1234", password_confirmation: "secret1234"}

  test "lock/2" do
    assert {:ok, user} = Repo.insert(User.changeset(%User{}, @valid_params))
    refute user.locked_at

    assert {:ok, user} = Users.lock(user)
    assert user.locked_at

    assert {:error, changeset} = Users.lock(user)
    assert changeset.errors[:locked_at] == {"already set", []}
  end
end
# test/my_app_web/controllers/admin/user_controller_test.exs
defmodule MyAppWeb.Admin.UserControllerTest do
  use MyAppWeb.ConnCase

  alias MyApp.{Users, Users.User, Repo}

  describe "lock/2" do
    test "locks user", %{conn: conn} do
      user = user_fixture()

      conn = post(conn, ~p"/admin/users/#{user.id}/lock")

      assert Phoenix.Flash.get(conn.assigns.flash, :info) == "User has been locked."
      assert redirected_to(conn) == "/"
    end

    test "with already locked user", %{conn: conn} do
      {:ok, user} = Users.lock(user_fixture())

      conn = post(conn, ~p"/admin/users/#{user.id}/lock")

      assert Phoenix.Flash.get(conn.assigns.flash, :error) == "User couldn't be locked."
      assert redirected_to(conn) == "/"
    end
  end

  defp user_fixture() do
    %User{}
    |> User.changeset(%{email: "test@example.com", password: "secret1234", password_confirmation: "secret1234"})
    |> Repo.insert!()
  end
end
# test/my_app_web/ensure_user_not_locked_plug_test.exs
defmodule MyAppWeb.EnsureUserNotLockedPlugTest do
  use MyAppWeb.ConnCase

  alias MyApp.Users.User
  alias MyAppWeb.EnsureUserNotLockedPlug

  @pow_config [otp_app: :my_app]
  @user %User{id: 1, locked_at: nil}
  @locked_user %User{id: 2, locked_at: DateTime.utc_now()}
  @plug_opts []

  setup do
    {:ok, conn: init_conn()}
  end

  test "call/2 with no user", %{conn: conn} do
    opts = EnsureUserNotLockedPlug.init(@plug_opts)
    conn = EnsureUserNotLockedPlug.call(conn, opts)

    refute conn.halted
  end

  test "call/2 with user", %{conn: conn} do
    opts = EnsureUserNotLockedPlug.init(@plug_opts)
    conn =
      conn
      |> Pow.Plug.assign_current_user(@user, @pow_config)
      |> EnsureUserNotLockedPlug.call(opts)

    refute conn.halted
  end

  test "call/2 with locked user", %{conn: conn} do
    opts = EnsureUserNotLockedPlug.init(@plug_opts)
    conn =
      conn
      |> Pow.Plug.assign_current_user(@locked_user, @pow_config)
      |> EnsureUserNotLockedPlug.call(opts)

    assert Phoenix.Flash.get(conn.assigns.flash, :error) == "Sorry, your account is locked."
    assert redirected_to(conn) == ~p"/session/new"
  end

  defp init_conn() do
    pow_config = Keyword.put(@pow_config, :plug, Pow.Plug.Session)

    :get
    |> Plug.Test.conn("/")
    |> Plug.Test.init_test_session(%{})
    |> Pow.Plug.put_config(pow_config)
    |> fetch_flash()
  end
end
# test/my_app_web/controllers/reset_password_controller_test.exs
defmodule MyAppWeb.ResetPasswordControllerTest do
  use MyAppWeb.ConnCase

  alias MyApp.{Users, Users.User, Repo}
  alias PowResetPassword.Store.ResetTokenCache

  describe "create/2" do
    @valid_params %{"user" => %{"email" => "test@example.com"}}

    test "with user", %{conn: conn} do
      user = user_fixture()

      conn = post(conn, ~p"/reset-password", @valid_params)

      assert Phoenix.Flash.get(conn.assigns.flash, :info)
      assert redirected_to(conn) == ~p"/session/new"

      assert count_reset_password_tokens_for_user(conn, user) == 1
    end

    test "with locked user", %{conn: conn} do
      {:ok, user} = Users.lock(user_fixture())

      conn = post(conn, ~p"/reset-password", @valid_params)

      assert Phoenix.Flash.get(conn.assigns.flash, :info)
      assert redirected_to(conn) == ~p"/session/new"

      assert count_reset_password_tokens_for_user(conn, user) == 0
    end
  end

  defp user_fixture() do
    %User{}
    |> User.changeset(%{email: "test@example.com", password: "secret1234", password_confirmation: "secret1234"})
    |> Repo.insert!()
  end

  defp count_reset_password_tokens_for_user(conn, user) do
    backend =
      conn
      |> Pow.Plug.fetch_config()
      |> Pow.Config.get(:cache_store_backend, Pow.Store.Backend.EtsCache)

    [backend: backend]
    |> ResetTokenCache.all([:_])
    |> Enum.filter(fn {_key, %{id: id}} -> id == user.id end)
    |> length()
  end
end


  

    
Custom controllers
    

Please note that if you just wish to modify the templates, then you should follow the Modify templates section(s) in the README. This guide is meant for allowing complete control over flow.
Pow makes it easy to use custom controllers leveraging the underlying Pow logic. It is ideal for cases where you need to control the flow, e.g., protect the registration process in a certain way.
First you should follow the Getting Started section in README until before the router.ex modification.

  
    
  
  Routes


Modify your WEB_PATH/router.ex to set up the Pow plugs in :protected and :not_authenticated pipelines with a custom error handler, and add the routes for your custom session and registration controllers:
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  # ... pipelines

  pipeline :protected do
    plug Pow.Plug.RequireAuthenticated,
      error_handler: MyAppWeb.AuthErrorHandler
  end

  pipeline :not_authenticated do
    plug Pow.Plug.RequireNotAuthenticated,
      error_handler: MyAppWeb.AuthErrorHandler
  end

  scope "/", MyAppWeb do
    pipe_through [:browser, :not_authenticated]

    get "/signup", RegistrationController, :new, as: :signup
    post "/signup", RegistrationController, :create, as: :signup
    get "/login", SessionController, :new, as: :login
    post "/login", SessionController, :create, as: :login
  end

  scope "/", MyAppWeb do
    pipe_through [:browser, :protected]

    delete "/logout", SessionController, :delete, as: :logout
  end

  # ... routes
end
And create WEB_PATH/auth_error_handler.ex:
defmodule MyAppWeb.AuthErrorHandler do
  use MyAppWeb, :controller
  alias Plug.Conn

  @spec call(Conn.t(), atom()) :: Conn.t()
  def call(conn, :not_authenticated) do
    conn
    |> put_flash(:error, "You've to be authenticated first")
    |> redirect(to: ~p"/login")
  end

  @spec call(Conn.t(), atom()) :: Conn.t()
  def call(conn, :already_authenticated) do
    conn
    |> put_flash(:error, "You're already authenticated")
    |> redirect(to: ~p"/")
  end
end
This module will make sure that unauthenticated user can't log out, and authenticated users can't sign in again.

  
    
  
  Controllers


We'll be using Pow.Plug for the heavy lifting, and customizing the response handling in our controllers.
Create WEB_PATH/controllers/registration_controller.ex:
defmodule MyAppWeb.RegistrationController do
  use MyAppWeb, :controller

  def new(conn, _params) do
    # We'll leverage `Pow.Plug`, but you can also follow the classic Phoenix way:
    # changeset = MyApp.Users.User.changeset(%MyApp.Users.User{}, %{})

    changeset = Pow.Plug.change_user(conn)

    render(conn, "new.html", changeset: changeset)
  end

  def create(conn, %{"user" => user_params}) do
    # We'll leverage `Pow.Plug`, but you can also follow the classic Phoenix way:
    # user =
    #   %MyApp.Users.User{}
    #   |> MyApp.Users.User.changeset(user_params)
    #   |> MyApp.Repo.insert()

    conn
    |> Pow.Plug.create_user(user_params)
    |> case do
      {:ok, user, conn} ->
        conn
        |> put_flash(:info, "Welcome!")
        |> redirect(to: ~p"/")

      {:error, changeset, conn} ->
        render(conn, "new.html", changeset: changeset)
    end
  end
end
Create WEB_PATH/controllers/session_controller.ex:
defmodule MyAppWeb.SessionController do
  use MyAppWeb, :controller

  def new(conn, _params) do
    changeset = Pow.Plug.change_user(conn)

    render(conn, "new.html", changeset: changeset)
  end

  def create(conn, %{"user" => user_params}) do
    conn
    |> Pow.Plug.authenticate_user(user_params)
    |> case do
      {:ok, conn} ->
        conn
        |> put_flash(:info, "Welcome back!")
        |> redirect(to: ~p"/")

      {:error, conn} ->
        changeset = Pow.Plug.change_user(conn, conn.params["user"])

        conn
        |> put_flash(:info, "Invalid email or password")
        |> render("new.html", changeset: changeset)
    end
  end

  def delete(conn, _params) do
    conn
    |> Pow.Plug.delete()
    |> redirect(to: ~p"/")
  end
end

  
    
  
  Templates


Create WEB_PATH/controllers/registration_html/new.html.heex:
<div class="mx-auto max-w-sm">
  <.header class="text-center">
    Register account
    <:subtitle>
      Already registered?
      <.link navigate={~p"/login"} class="font-semibold text-brand hover:underline">
        Log in
      </.link>
      instead.
    </:subtitle>
  </.header>

  <.simple_form :let={f} for={@changeset} action={~p"/signup"}>
    <.error :if={@changeset.action == :insert}>
      Oops, something went wrong! Please check the errors below.
    </.error>

    <.input field={f[:email]} type="email" label="Email" required />
    <.input field={f[:password]} type="password" label="Password" required />
    <.input field={f[:password_confirmation]} type="password" label="Confirm password" required />

    <:actions>
      <.button phx-disable-with="Registering account..." class="w-full">Register</.button>
    </:actions>
  </.simple_form>
</div>
Create WEB_PATH/controllers/session_html/new.html.heex:
<div class="mx-auto max-w-sm">
  <.header class="text-center">
    Sign in
    <:subtitle>
      No account?
      <.link navigate={~p"/signup"} class="font-semibold text-brand hover:underline">
        Register
      </.link>
      instead.
    </:subtitle>
  </.header>

  <.simple_form :let={f} for={@changeset} action={~p"/login"}>
    <.error :if={@changeset.action == :insert}>
      Oops, something went wrong! Please check the errors below.
    </.error>

    <.input field={f[:email]} type="email" label="Email" required />
    <.input field={f[:password]} type="password" label="Password" required />

    <:actions>
      <.button phx-disable-with="Registering account..." class="w-full">Register</.button>
    </:actions>
  </.simple_form>
</div>
Remember to create the template files WEB_PATH/controllers/registration_html.ex and WEB_PATH/controllers/session_html.ex too.

  
    
  
  Further customization


That's all you need to do to have custom controllers with Pow! From here on you can customize your flow.
You may want to utilize some of the extensions, but since you have created a custom controller, it's highly recommended that you do not rely on any controller actions in the extensions. Instead, you should implement the logic yourself to keep your controllers as explicit as possible. This is only an example:
defmodule MyAppWeb.SessionController do
  # ...

  def create(conn, %{"user" => user_params}) do
    conn
    |> Pow.Plug.authenticate_user(user_params)
    |> verify_confirmed()
  end

  defp verify_confirmed({:ok, conn}) do
    conn
    |> Pow.Plug.current_user()
    |> email_confirmed?()
    |> case do
      true ->
        conn
        |> put_flash(:info, "Welcome back!")
        |> redirect(to: ~p"/")

      false ->
        conn
        |> Pow.Plug.delete()
        |> put_flash(:info, "Your e-mail address has not been confirmed.")
        |> redirect(to: ~p"/login")
    end
  end
  defp verify_confirmed({:error, conn}) do
    changeset = Pow.Plug.change_user(conn, conn.params["user"])

    conn
    |> put_flash(:info, "Invalid email or password")
    |> render("login.html", changeset: changeset)
  end

  defp email_confirmed?(%{email_confirmed_at: nil, email_confirmation_token: token, unconfirmed_email: nil}) when not is_nil(token), do: false
  defp email_confirmed?(_user), do: true
  # ...
end


  

    
Disable registration
    

You may have an app in which users are not permitted to sign up. Pow makes it easy to disable registration by removing the registration routes.
First you should follow the Modify templates section in README.

  
    
  
  Templates


Open up WEB_PATH/controllers/pow/session_html/new.html.heex and remove the registration link. Delete the WEB_PATH/controllers/pow/registration_html/new.html.heex file.

  
    
  
  Routes


Replace pow_routes() with pow_session_routes() in your router module.
Add the following routes below to enable account updates and deletion:
scope "/", Pow.Phoenix, as: "pow" do
  pipe_through [:browser, :protected]
  resources "/registration", RegistrationController, singleton: true, only: [:edit, :update, :delete]
end
That's it!


  

    
Redis cache store backend
    

For read-only systems, like Heroku, you won't be able to use the built-in Mnesia backend cache for distribution and to persist cache data between restarts. Instead, let's use Redix to store our cache data in Redis.
First, add Redix to your list of dependencies in mix.exs:
# mix.exs
defp deps do
  [
    # ...
    {:redix, "~> 0.9.2"}
  ]
end
Now set up your WEB_PATH/pow/redis_cache.ex like so:
# lib/my_app_web/pow/redis_cache.ex
defmodule MyAppWeb.Pow.RedisCache do
  @moduledoc """
  Redis cache with optimized lookup.

  To keep lookups performant the key is split up into n - 1 prefixes, where n
  is the number of items in the key array. Each prefix will be combined with
  all previous prefixes (if any) into a sorted set key. The rest of the key
  will be combined into a member of this sorted set. The member of the sorted
  set will have the expiration timestamp set as the score.

  The sorted set will be set to expire with the provided `ttl`. When new
  records are inserted any members of any of these prefix sorted sets with a
  score score lower than the current timestamp will be removed.

  When records are queried with `all/2`, the first matching `:_` wildcard in
  the key match will define what prefix sorted set to use to find the keys. All
  members with a lower score than the current timestamp will be removed, and
  the sorted set for that prefix will be scanned. The resulting list is then
  checked against the key match spec.

  Any part of the key and value will be encoded with `:erlang.term_to_binary/1`
  and `Base.url_encode64/2` when stored.

  ## Configuration options

    * `:ttl` - integer value in milliseconds for ttl of records. If this value
      is not provided, or is set to nil, the records will never expire.

    * `:namespace` - value to use for namespacing keys. Defaults to "cache".

    * `:writes` - set to `:async` to do asynchronous writes. Defaults to
      `:sync`.
  """

  @behaviour Pow.Store.Backend.Base

  alias Pow.Config

  @redix_instance_name :redix

  @impl true
  def put(config, record_or_records) do
    ttl = Config.get(config, :ttl) || raise_ttl_error!()

    commands =
      record_or_records
      |> List.wrap()
      |> Enum.reduce([], fn {key, value}, acc ->
        config
        |> redis_key(key)
        |> command_builder(&put_command(&1, value, ttl))
        |> flatten(acc)
      end)

    maybe_async(config, fn ->
      @redix_instance_name
      |> Redix.pipeline!(commands)
      |> Enum.zip(commands)
      |> Enum.reject(fn
        {n, _cmd} when is_number(n) or n == "OK" -> true
        _any -> false
      end)
      |> case do
        []     -> :ok
        errors -> raise "Redix received unexpected response #{inspect errors}"
      end
    end)

    :ok
  end

  defp put_command({key, []}, value, ttl) do
    key   = to_binary_redis_key(key)
    value = :erlang.term_to_binary(value)

    ["SET", key, value, "PX", ttl]
  end
  defp put_command({prefix, key}, _value, ttl) do
    key       = to_binary_redis_key(key)
    timestamp = current_timestamp()

    index_key =
      prefix
      |> to_binary_redis_key()
      |> to_index_key()

    [
      ["ZREMRANGEBYSCORE", index_key, "-inf", timestamp],
      ["ZADD", index_key, timestamp + ttl, key],
      ["PEXPIRE", index_key, ttl]
    ]
  end

  defp current_timestamp, do: DateTime.to_unix(DateTime.utc_now(), :millisecond)

  defp command_builder(key, fun) do
    count = Enum.count(key)

    1..count
    |> Enum.map(fn i ->
      key
      |> Enum.split(i)
      |> fun.()
    end)
    |> Enum.reduce([], &flatten/2)
  end

  defp flatten([item | _rest] = items, acc) when is_list(item)  do
    Enum.reduce(items, acc, &flatten/2)
  end
  defp flatten(item, acc), do: acc ++ [item]

  defp maybe_async(config, fun) do
    case Config.get(config, :writes, :sync) do
      :sync -> fun.()
      :async -> Task.start(fun)
    end
  end

  @impl true
  def delete(config, key) do
    commands =
      config
      |> redis_key(key)
      |> command_builder(&delete_command/1)

    maybe_async(config, fn ->
      Redix.pipeline!(@redix_instance_name, commands)
    end)

    :ok
  end

  def delete_command({prefix, []}) do
    prefix = to_binary_redis_key(prefix)

    ["DEL", prefix]
  end
  def delete_command({prefix, key}) do
    index_key =
      prefix
      |> to_binary_redis_key()
      |> to_index_key()

    key = to_binary_redis_key(key)

    ["ZREM", index_key, key]
  end

  @impl true
  def get(config, key) do
    key =
      config
      |> redis_key(key)
      |> to_binary_redis_key()

    case Redix.command!(@redix_instance_name, ["GET", key]) do
      nil   -> :not_found
      value -> :erlang.binary_to_term(value)
    end
  end

  @impl true
  def all(config, key_match) do
    compiled_match_spec = :ets.match_spec_compile([{{key_match, :_}, [], [:"$_"]}])
    key_match           = redis_key(config, key_match)

    prefix =
      key_match
      |> Enum.find_index(& &1 == :_)
      |> case do
        nil -> {redis_key(config, []), key_match}
        i   -> Enum.split(key_match, i)
      end
      |> elem(0)

    index_key =
      prefix
      |> to_binary_redis_key()
      |> to_index_key()

    Redix.command!(@redix_instance_name, ["ZREMRANGEBYSCORE", index_key, "-inf", current_timestamp()])

    Stream.resource(
      fn -> do_scan(config, prefix, compiled_match_spec, "0") end,
      &stream_scan(config, prefix, compiled_match_spec, &1),
      fn _ -> :ok end)
    |> Enum.to_list()
  end

  defp to_index_key(key), do: "_index:#{key}"

  defp stream_scan(_config, _prefix, _compiled_match_spec, {[], "0"}), do: {:halt, nil}
  defp stream_scan(config, prefix, compiled_match_spec, {[], iterator}) do
    result = do_scan(config, prefix, compiled_match_spec, iterator)

    stream_scan(config, prefix, compiled_match_spec, result)
  end
  defp stream_scan(_config, _prefix, _compiled_match_spec, {keys, iterator}), do: {keys, {[], iterator}}

  defp do_scan(config, prefix, compiled_match_spec, iterator) do
    prefix    = to_binary_redis_key(prefix)
    index_key = to_index_key(prefix)

    [iterator, res] = Redix.command!(@redix_instance_name, ["ZSCAN", index_key, iterator])

    keys = Enum.take_every(res, 2)

    {filter_or_load_value(compiled_match_spec, prefix, keys, config), iterator}
  end

  defp filter_or_load_value(compiled_match_spec, prefix, keys, config) do
    keys
    |> Enum.map(&"#{prefix}:#{&1}")
    |> Enum.map(&convert_key/1)
    |> :ets.match_spec_run(compiled_match_spec)
    |> populate_values(config)
  end

  defp convert_key(key) do
    key =
      key
      |> from_binary_redis_key()
      |> unwrap()

    {key, nil}
  end

  defp unwrap([_namespace, key]), do: key
  defp unwrap([_namespace | key]), do: key

  defp populate_values([], _config), do: []
  defp populate_values(records, config) do
    binary_keys = Enum.map(records, fn {key, nil} -> binary_redis_key(config, key) end)

    values =
      @redix_instance_name
      |> Redix.command!(["MGET"] ++ binary_keys)
      |> Enum.map(fn
        nil   -> nil
        value -> :erlang.binary_to_term(value)
      end)

    records
    |> zip_values(values)
    |> Enum.reject(fn {_key, value} -> is_nil(value) end)
  end

  defp zip_values([{key, nil} | next1], [value | next2]) do
    [{key, value} | zip_values(next1, next2)]
  end
  defp zip_values(_, []), do: []
  defp zip_values([], _), do: []

  defp binary_redis_key(config, key) do
    config
    |> redis_key(key)
    |> to_binary_redis_key()
  end

  defp redis_key(config, key) do
    [namespace(config) | List.wrap(key)]
  end

  defp namespace(config), do: Config.get(config, :namespace, "cache")

  defp to_binary_redis_key(key) do
    key
    |> Enum.map(fn part ->
      part
      |> :erlang.term_to_binary()
      |> Base.url_encode64(padding: false)
    end)
    |> Enum.join(":")
  end

  defp from_binary_redis_key(key) do
    key
    |> String.split(":")
    |> Enum.map(fn part ->
      part
      |> Base.url_decode64!(padding: false)
      |> :erlang.binary_to_term()
    end)
  end

  @spec raise_ttl_error!() :: no_return()
  defp raise_ttl_error!(),
    do: Config.raise_error("`:ttl` configuration option is required for #{inspect(__MODULE__)}")
end
We are converting keys to binary keys since we can't directly use the Erlang terms as with ETS and Mnesia.
We'll need to start the Redix application on our app startup, so in application.ex add {Redix, name: :redix} to your supervision tree:
# lib/my_app/application.ex
defmodule MyApp.Application do

  # ...

  def start(_type, _args) do
    import Supervisor.Spec

    children = [
      # ...
      {Redix, name: :redix}
    ]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end

  # ...
end
By default localhost Redis is used, but you can update this by using a Redis URI: {Redix, {"redis://:secret@redix.example.com:6380/1", [name: :redix]}}
Finally update the config with your new Redis cache backend:
# config/config.ex
config :my_app, :pow,
  user: MyApp.Users.User,
  repo: MyApp.Repo,
  cache_store_backend: MyAppWeb.Pow.RedisCache
And now you have a running Redis cache store backend!

  
    
  
  Test module


# test/my_app_web/pow/redis_cache_test.exs
defmodule MyAppWeb.Pow.RedisCacheTest do
  use ExUnit.Case
  doctest MyAppWeb.Pow.RedisCache

  alias MyAppWeb.Pow.RedisCache

  @default_config [namespace: "test", ttl: :timer.hours(1)]

  setup do
    Redix.command!(:redix, ["FLUSHALL"])

    :ok
  end

  test "can put, get and delete records" do
    assert RedisCache.get(@default_config, "key") == :not_found

    RedisCache.put(@default_config, {"key", "value"})
    assert RedisCache.get(@default_config, "key") == "value"

    RedisCache.delete(@default_config, "key")
    assert RedisCache.get(@default_config, "key") == :not_found
  end

  test "with `writes: :async` config option" do
    config = Keyword.put(@default_config, :writes, :async)

    assert RedisCache.get(config, "key") == :not_found

    RedisCache.put(config, {"key", "value"})
    assert RedisCache.get(config, "key") == :not_found
    :timer.sleep(100)
    assert RedisCache.get(config, "key") == "value"

    RedisCache.delete(config, "key")
    assert RedisCache.get(config, "key") == "value"
    :timer.sleep(100)
    assert RedisCache.get(config, "key") == :not_found
  end

  test "delete removes from redis sets" do
    RedisCache.put(@default_config, {["namespace", "key1"], "value"})
    RedisCache.put(@default_config, {["namespace", "key2"], "value"})
    assert redix_zmembers_decoded("namespace") == ["key1", "key2"]

    RedisCache.delete(@default_config, ["namespace", "key1"])
    assert redix_zmembers_decoded("namespace") == ["key2"]
  end

  describe "with redis errors" do
    setup do
      ["maxmemory", value] = Redix.command!(:redix, ["CONFIG", "GET", "maxmemory"])

      Redix.command!(:redix, ["CONFIG", "SET", "maxmemory", "10"])

      on_exit(fn ->
        Redix.command!(:redix, ["CONFIG", "SET", "maxmemory", value])
      end)
    end

    test "raises error" do
      expected_error_message =
        "Redix received unexpected response [{%Redix.Error{message: \"OOM command not allowed when used memory > 'maxmemory'.\"}, "

      assert_raise RuntimeError, ~r/#{Regex.escape(expected_error_message)}/, fn ->
        RedisCache.put(@default_config, {"key", "value"})
      end
    end
  end

  test "can put multiple records at once" do
    RedisCache.put(@default_config, [{"key1", "1"}, {"key2", "2"}])
    assert RedisCache.get(@default_config, "key1") == "1"
    assert RedisCache.get(@default_config, "key2") == "2"
  end

  test "can match fetch all" do
    assert RedisCache.all(@default_config, :_) == []

    for number <- 1..11, do: RedisCache.put(@default_config, {"key#{number}", "value"})
    items = RedisCache.all(@default_config, :_)

    assert Enum.find(items, fn {key, "value"} -> key == "key1" end)
    assert Enum.find(items, fn {key, "value"} -> key == "key2" end)
    assert length(items) == 11

    RedisCache.put(@default_config, {["namespace", "key"], "value"})
    RedisCache.put(@default_config, {["namespace", "key", "key2"], "value"})
    assert RedisCache.all(@default_config, ["namespace", :_]) == [{["namespace", "key"], "value"}]
  end

  test "records auto purge" do
    config = Keyword.put(@default_config, :ttl, 50)

    RedisCache.put(config, {"key", "value"})
    RedisCache.put(config, [{"key1", "1"}, {["namespace", "key2"], "2"}])
    assert RedisCache.get(config, "key") == "value"
    assert RedisCache.get(config, "key1") == "1"
    assert RedisCache.get(config, ["namespace", "key2"]) == "2"
    assert redix_zmembers_decoded("namespace") == ["key2"]
    RedisCache.put(Keyword.put(@default_config, :ttl, 150), [{["namespace", "key3"], "3"}])
    assert redix_zmembers_decoded("namespace") == ["key2", "key3"]
    :timer.sleep(100)
    assert RedisCache.get(config, "key") == :not_found
    assert RedisCache.get(config, "key1") == :not_found
    assert RedisCache.get(config, ["namespace", "key2"]) == :not_found
    assert RedisCache.get(config, ["namespace", "key3"]) == "3"
    RedisCache.put(Keyword.put(@default_config, :ttl, 100), [{["namespace", "key4"], "4"}])
    assert redix_zmembers_decoded("namespace") == ["key3", "key4"]
    :timer.sleep(100)
    assert RedisCache.get(config, ["namespace", "key3"]) == :not_found
    assert RedisCache.all(config, ["namespace", :_]) ==  []
    assert redix_zmembers_decoded("namespace") == []
  end

  defp redix_zmembers_decoded(key) do
    encoded_prefix =
      [@default_config[:namespace]]
      |> Kernel.++(List.wrap(key))
      |> Enum.map(fn part ->
        part
        |> :erlang.term_to_binary()
        |> Base.url_encode64(padding: false)
      end)
      |> Enum.join(":")

    :redix
    |> Redix.command!(["ZRANGEBYSCORE", "_index:#{encoded_prefix}", "-inf", "+inf"])
    |> Enum.map(fn encoded_key ->
      encoded_key
      |> String.split(":")
      |> Enum.map(fn part ->
        part
        |> Base.url_decode64!(padding: false)
        |> :erlang.binary_to_term()
      end)
      |> case do
        [key] -> key
        key   -> key
      end
    end)
  end
end


  

    
Pow in an umbrella project
    

Adding Pow to an umbrella project is near identical to a single app setup.
Instead of running mix pow.install, you should run mix pow.ecto.install inside your ecto app, and mix pow.phoenix.install inside your phoenix app(s).
You can follow the rest of the README instructions with the following caveats in mind:
	Run all mix ecto tasks inside the ecto app
	Run all mix phoenix tasks inside the phoenix app(s)
	For a project generated with mix phx.new --umbrella use :my_app_web instead of :my_app



  

    
Multitenancy with Pow
    

You can pass repo options to the functions used in Pow.Ecto.Context by using the :repo_opts configuration option. This makes it possible to pass on the prefix option used in multitenancy apps, so you can do the following:
config :my_app, :pow,
  # ...
  repo_opts: [prefix: "tenant_a"]
You can also pass the prefix option to Pow.Plug.Session in your WEB_PATH/endpoint.ex:
plug Pow.Plug.Session, otp_app: :my_app, repo_opts: [prefix: "tenant_a"]
And you can add it as a custom plug to use a dynamic prefix value (in this case the conn.private[:tenant_prefix] has been set beforehand):
defmodule MyAppWeb.Pow.TenantPlug do
  def init(config), do: config

  def call(conn, config) do
    prefix = conn.private[:tenant_prefix]
    config = Keyword.put(config, :repo_opts, [prefix: prefix])

    Pow.Plug.Session.call(conn, config)
  end
end

  
    
  
  Process dictionary


Another common approach is to use the process dictionary. In that case you won't have to pass any repo opts. You just use your modified MyApp.Repo module to set and fetch the tenant id in a controller or plug. It could look like:
defmodule MyAppWeb.SetTenantPlug do
  def init(opts), do: opts

  def call(conn, _opts_) do
    MyApp.Repo.put_org_id(conn.private[:tenant_org_id])

    conn
  end
end

  
    
  
  Triplex


Triplex is a database multitenancy package. Pow can be easily configured to work with Triplex.
Update your WEB_PATH/endpoint.ex using a custom plug rather than the default Pow.Plug.Session:
# lib/my_app_web/endpoint.ex
defmodule MyAppWeb.Endpoint do
  use Phoenix.Endpoint, otp_app: :my_app

  # ...

  # You should load the tenant with Triplex before calling the
  # `TriplexSessionPlug`. If you use the `ParamPlug` you could add it here:
  # plug Triplex.ParamPlug, param: :subdomain

  # ...

  plug Plug.Session, @session_options
  plug MyAppWeb.Pow.TriplexSessionPlug, otp_app: :my_app
  # ...
end
Then set up WEB_PATH/pow/triplex_session_plug.ex:
# lib/my_app_web/pow/triplex_session_plug.ex
defmodule MyAppWeb.Pow.TriplexSessionPlug do
  def init(config), do: config

  def call(conn, config) do
    tenant = conn.assigns[:current_tenant] || conn.assigns[:raw_current_tenant]
    prefix = Triplex.to_prefix(tenant)
    config = Keyword.put(config, :repo_opts, [prefix: prefix])

    Pow.Plug.Session.call(conn, config)
  end
end

  
    
  
  Triplex create tenant migration with user


You may want to create an new tenant with a Pow user as part of account registration.
First we'll add the account context module that will create an account with a user using an Ecto.Multi transaction:
# test/my_app/accounts.exs
defmodule MyApp.Accounts do
  alias Ecto.Changeset
  alias MyApp.{Repo, Users.User}

  @otp_app :my_app
  @tenant_param :tenant

  def create_account(tenant_id, params) do
    tenant_id
    |> Triplex.create_schema(Repo, fn (tenant, repo) ->
      do_create_account(tenant, repo, params)
    end)
    |> case do
      {:ok, tenant} -> {:ok, Repo.one!(User, prefix: tenant)}
      {:error, %Changeset{} = changeset} -> {:error, changeset}
      {:error, reason} -> invalid_tenant_changeset_error(params, "couldn't be created", reason)
    end
  end

  defp do_create_account(tenant, repo, params) do
    pow_config =
      [
        otp_app: @otp_app,
        repo: repo,
        user: User,
        repo_opts: [prefix: tenant]
      ]

    Ecto.Multi.new()
    |> Ecto.Multi.run(:triplex_migration, fn repo, %{} ->
      Triplex.migrate(tenant, repo)
    end)
    |> Ecto.Multi.run(:user, fn _, _ ->
      Pow.Ecto.Context.create(params, pow_config)
    end)
    |> repo.transaction()
    |> case do
      {:ok, any} ->
        {:ok, any}

      {:error, :triplex_migration, reason, _} ->
        invalid_tenant_changeset_error(params, "couldn't be created", [reason: reason])

      {:error, :user, changeset, _} ->
        {:error, changeset}
    end
  end

  defp invalid_tenant_changeset_error(params, error, keys) do
    changeset =
      %User{}
      |> User.changeset(params)
      |> Changeset.add_error(@tenant_param, error, keys)

    {:error, changeset}
  end
end
Next we'll add the controller:
# lib/my_app_web/controllers/account_controller.ex
defmodule MyAppWeb.AccountController do
  use MyAppWeb, :controller

  alias MyApp.Accounts

  def new(conn, _params) do
    render(conn, "new.html")
  end

  def create(conn, %{"account" => tenant_id, "user" => user_params}) do
    case Accounts.create_account(tenant_id, user_params) do
      {:ok, user} ->
        conn
        |> Pow.Plug.create(user)
        |> put_flash(:info, "Welcome!")
        |> redirect(to: ~p"/")

      {:error, changeset} ->
        render(conn, "new.html", changeset: changeset)
    end
  end
end
Now all you need is to set up the template, and update your router module.

  
    
  
  Triplex test modules


# lib/my_app_web/pow/triplex_session_plug_test.exs
defmodule MyAppWeb.Pow.TriplexSessionPlugTest do
  use MyAppWeb.ConnCase

  alias MyAppWeb.Pow.TriplexSessionPlug

  @pow_config [otp_app: :my_app]
  @tenant_a "tenant_a"
  @tenant_b "tenant_b"
  @valid_user_params %{
    "email" => "test@example.com",
    "password" => "password",
    "password_confirmation" => "password"
  }

  setup do
    {:ok, conn: init_conn()}
  end

  test "handles Triplex tenants", %{conn: conn} do
    opts = TriplexSessionPlug.init(@pow_config)
    {:ok, _user, _conn} =
      conn
      |> set_triplex_tenant(@tenant_a)
      |> TriplexSessionPlug.call(opts)
      |> Pow.Plug.create_user(@valid_user_params)

    {:error, _conn} =
      conn
      |> set_triplex_tenant(@tenant_b)
      |> TriplexSessionPlug.call(opts)
      |> Pow.Plug.authenticate_user(@valid_user_params)

    {:ok, _conn} =
      conn
      |> set_triplex_tenant(@tenant_a)
      |> TriplexSessionPlug.call(opts)
      |> Pow.Plug.authenticate_user(@valid_user_params)
  end

  defp init_conn() do
    :get
    |> Plug.Test.conn("/")
    |> Plug.Test.init_test_session(%{})
    |> fetch_flash()
  end

  defp set_triplex_tenant(conn, tenant) do
    conn = %{conn | params: %{"subdomain" => tenant}}
    opts = Triplex.ParamPlug.init(param: :subdomain)

    Triplex.ParamPlug.call(conn, opts)
  end
end
# test/my_app/accounts_test.exs
defmodule MyApp.AccountsTest do
  use MyApp.DataCase

  alias MyApp.{Repo, Accounts}

  @tenant_id "test_tenant"
  @tenant_param :tenant
  @valid_params %{email: "test@example.com", password: "secret1234", password_confirmation: "secret1234"}

  setup do
    Ecto.Adapters.SQL.Sandbox.mode(Repo, :auto)

    Triplex.drop(@tenant_id, Repo)

    on_exit fn ->
      Triplex.drop(@tenant_id, Repo)
    end

    :ok
  end

  test "create_account/2" do
    assert {:ok, user} = Accounts.create_account(@tenant_id, @valid_params)
    assert user.__meta__.prefix == @tenant_id

    assert {:error, changeset} = Accounts.create_account(@tenant_id, @valid_params)
    assert changeset.errors[@tenant_param] == {"couldn't be created", "ERROR 42P06 (duplicate_schema) schema \"#{@tenant_id}\" already exists"}
  end
end
# test/my_app_web/controllers/account_controller_test.exs
defmodule MyAppWeb.AccountControllerTest do
  use MyAppWeb.ConnCase

  alias MyApp.Repo

  @tenant_id "test-tenant"

  setup do
    Ecto.Adapters.SQL.Sandbox.mode(Repo, :auto)

    Triplex.drop(@tenant_id, Repo)

    on_exit fn ->
      Triplex.drop(@tenant_id, Repo)
    end

    :ok
  end

  describe "create/2" do
    @valid_params %{"account" => @tenant_id, "user" => %{"email" => "test@example.com", "password" => "secret1234", "password_confirmation" => "secret1234"}}
    @invalid_params %{"account" => @tenant_id, "user" => %{"email" => "test@example.com", "password" => "secret1234"}}

    test "with invalid params", %{conn: conn} do
      conn = post(conn, ~p"/account", @invalid_params)

      assert html_response(conn, 500)
      refute Pow.Plug.current_user(conn)
    end

    test "with valid params", %{conn: conn} do
      conn = post(conn, ~p"/account", @valid_params)

      assert Phoenix.Flash.get(conn.assigns.flash, :info) == "Welcome!"
      assert redirected_to(conn) == ~p"/"

      assert Pow.Plug.current_user(conn)
    end
  end
end


  

    
Update cached user credentials
    

You may want to update the cached user credentials when an action outside of Pow has updated the user. It's very important to understand that the cached user credentials that Pow fetches in Pow.Plug.current_user/2 is always to be considered out of date since it's a cached object.
In the following examples, we'll imagine that you've added a plan column on your users table. We may want to use that plan to give them access to certain controller actions. In this case, it's paramount that you load the user from the database.

  
    
  
  Update Pow.Store.CredentialsCache config


If you wish to load the user object each time it's fetched from the cache, all you have to do is to set reload: true for the Pow.Store.CredentialsCache config by adding this to your Pow config:
credentials_cache_store: {Pow.Store.CredentialsCache, reload: true}

  
    
  
  Reload the user with a plug


defmodule MyAppWeb.ProPlanController do
  # ...

  plug :reload_user

  # ...

  defp reload_user(conn, _opts) do
    config        = Pow.Plug.fetch_config(conn)
    user          = Pow.Plug.current_user(conn, config)
    reloaded_user = MyApp.Repo.get!(MyApp.User, user.id)

    Pow.Plug.assign_current_user(conn, reloaded_user, config)
  end
end
This should always be done for any authorization actions or any other actions that require the actual value to be known. Do note that only the controllers that have the plug will have the reloaded user. In all other controllers, the old cached credentials will be loaded instead.
If you would like to always fetch the user as it is in the database across all your controllers, you could instead set up a module plug and add it to your endpoint:
defmodule MyAppWeb.ReloadUserPlug do
  @doc false

  @spec init(any()) :: any()
  def init(opts), do: opts

  @doc false
  @spec call(Conn.t(), atom()) :: Conn.t()
  def call(conn, _opts) do
    config = Pow.Plug.fetch_config(conn)

    case Pow.Plug.current_user(conn, config) do
      nil ->
        conn

      user ->
        reloaded_user = MyApp.Repo.get!(MyApp.User, user.id)

        Pow.Plug.assign_current_user(conn, reloaded_user, config)
    end
  end
end
defmodule MyAppWeb.Endpoint do
  use Phoenix.Endpoint, otp_app: :my_app

  # ...

  plug Plug.Session, @session_options
  plug Pow.Plug.Session, otp_app: :my_app
  plug MyAppWeb.ReloadUserPlug
  # ...
end

  
    
  
  Update user in the credentials cache


Let's say that you want to show the user plan on most pages. In this case, we can safely rely on the cached credentials since we don't need to know the actual value in the database. The worst case is that a different plan may be shown if you haven't ensured that all plan update actions use the below function.
We can use Pow.Plug.create/2 to call the plug and update the cached credentials.
First, we'll make a helper and import it to our controllers:
defmodule MyAppWeb.Pow.Helper do
  @spec sync_user(Plug.Conn.t(), map()) :: Plug.Conn.t()
  def sync_user(conn, user), do: Pow.Plug.create(conn, user)
end
defmodule MyAppWeb do
  # ...

  def controller do
    quote do
      use Phoenix.Controller, namespace: MyAppWeb

      # ...
      import MyAppWeb.Pow.Helper
    end
  end

  # ...
end
Now we can call sync_user/2 in any controller actions. It could maybe be the update action for your plan controller:
defmodule MyAppWeb.PlanController do
  # ...

  def update(conn, %{"plan" => plan}) do
    conn
    |> Plug.current_user()
    |> MyApp.Users.update_plan(plan)
    |> case do
      {:ok, user} ->
        conn
        |> sync_user(user) # Update the user in the credentials cache
        |> put_flash(:info, "Plan updated successfully.")
        |> redirect(to: ~p"/profile")

      {:error, %Ecto.Changeset{} = changeset} ->
        render(conn, "plan.html", changeset: changeset)
    end
  end

  # ...
end
As you can see in the above, the cached user credentials will be updated after a successful update of the plan for the user. Now any subsequent pages being rendered, you'll have access to the updated plan value in the current user assign.
Another thing to note is that if you're using Pow.Plug.Session, then the session id will also be regenerated this way. This is ideal for authorization level change (what the above plan change action maybe).
You may also update the plan field in a background task. In this case, you won't have access to any current session, and you would have to use the Pow.Store.CredentialsCache.put/3 to update the credentials cache. However, since there are some caveats to this, it's instead recommended to find an alternative solution with the above functions.


  

    
How to use Pow in an API
    

Pow comes with plug n' play support for Phoenix as HTML web interface. API's work differently, and the developer should have full control over the flow in a proper built API. Therefore Pow encourages that you build custom controllers, and use the plug functions for API integration.
To get you started, here's the first steps to build a Pow enabled API interface.
We'll set up a custom authorization plug where we'll store session tokens with Pow.Store.CredentialsCache, and renewal tokens with PowPersistentSession.Store.PersistentSessionCache. The session tokens will automatically expire after 30 minutes, whereafter your client should request a new session token with the renewal token.
First you should follow the Getting Started section in README until before the WEB_PATH/endpoint.ex modification.

  
    
  
  Routes


Modify WEB_PATH/router.ex with API pipelines, and API endpoints for session and registration controllers:
# lib/my_app_web/router.ex
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  # # If you wish to also use Pow in your HTML frontend with session, then you
  # # should set the `Pow.Plug.Session` plug here rather than in the endpoint:
  # pipeline :browser do
  #   plug :accepts, ["html"]
  #   plug :fetch_session
  #   plug :fetch_flash
  #   plug Phoenix.LiveView.Flash
  #   plug :protect_from_forgery
  #   plug :put_secure_browser_headers
  #   plug Pow.Plug.Session, otp_app: :my_app
  # end

  pipeline :api do
    plug :accepts, ["json"]
    plug MyAppWeb.APIAuthPlug, otp_app: :my_app
  end

  pipeline :api_protected do
    plug Pow.Plug.RequireAuthenticated, error_handler: MyAppWeb.APIAuthErrorHandler
  end

  # ...

  scope "/api/v1", MyAppWeb.API.V1, as: :api_v1 do
    pipe_through :api

    resources "/registration", RegistrationController, singleton: true, only: [:create]
    resources "/session", SessionController, singleton: true, only: [:create, :delete]
    post "/session/renew", SessionController, :renew
  end

  scope "/api/v1", MyAppWeb.API.V1, as: :api_v1 do
    pipe_through [:api, :api_protected]

    # Your protected API endpoints here
  end

  # ... routes
end
As you can see, the above also shows how you can set up the browser pipeline in case you also have a web interface. You should put the Pow.Plug.Session plug there instead of in WEB_PATH/endpoint.ex.

  
    
  
  API authorization plug for Pow


Create WEB_PATH/api_auth_plug.ex with the following:
# lib/my_app_web/api_auth_plug.ex
defmodule MyAppWeb.APIAuthPlug do
  @moduledoc false
  use Pow.Plug.Base

  alias Plug.Conn
  alias Pow.{Config, Plug, Store.CredentialsCache}
  alias PowPersistentSession.Store.PersistentSessionCache

  @doc """
  Fetches the user from access token.
  """
  @impl true
  @spec fetch(Conn.t(), Config.t()) :: {Conn.t(), map() | nil}
  def fetch(conn, config) do
    with {:ok, signed_token} <- fetch_access_token(conn),
         {:ok, token}        <- verify_token(conn, signed_token, config),
         {user, _metadata}   <- CredentialsCache.get(store_config(config), token) do
      {conn, user}
    else
      _any -> {conn, nil}
    end
  end

  @doc """
  Creates an access and renewal token for the user.

  The tokens are added to the `conn.private` as `:api_access_token` and
  `:api_renewal_token`. The renewal token is stored in the access token
  metadata and vice versa.
  """
  @impl true
  @spec create(Conn.t(), map(), Config.t()) :: {Conn.t(), map()}
  def create(conn, user, config) do
    store_config  = store_config(config)
    access_token  = Pow.UUID.generate()
    renewal_token = Pow.UUID.generate()

    conn =
      conn
      |> Conn.put_private(:api_access_token, sign_token(conn, access_token, config))
      |> Conn.put_private(:api_renewal_token, sign_token(conn, renewal_token, config))
      |> Conn.register_before_send(fn conn ->
        # The store caches will use their default `:ttl` setting. To change the
        # `:ttl`, `Keyword.put(store_config, :ttl, :timer.minutes(10))` can be
        # passed in as the first argument instead of `store_config`.
        CredentialsCache.put(store_config, access_token, {user, [renewal_token: renewal_token]})
        PersistentSessionCache.put(store_config, renewal_token, {user, [access_token: access_token]})

        conn
      end)

    {conn, user}
  end

  @doc """
  Delete the access token from the cache.

  The renewal token is deleted by fetching it from the access token metadata.
  """
  @impl true
  @spec delete(Conn.t(), Config.t()) :: Conn.t()
  def delete(conn, config) do
    store_config = store_config(config)

    with {:ok, signed_token} <- fetch_access_token(conn),
         {:ok, token}        <- verify_token(conn, signed_token, config),
         {_user, metadata}   <- CredentialsCache.get(store_config, token) do

      Conn.register_before_send(conn, fn conn ->
        PersistentSessionCache.delete(store_config, metadata[:renewal_token])
        CredentialsCache.delete(store_config, token)

        conn
      end)
    else
      _any -> conn
    end
  end

  @doc """
  Creates new tokens using the renewal token.

  The access token, if any, will be deleted by fetching it from the renewal
  token metadata. The renewal token will be deleted from the store after the
  it has been fetched.
  """
  @spec renew(Conn.t(), Config.t()) :: {Conn.t(), map() | nil}
  def renew(conn, config) do
    store_config = store_config(config)

    with {:ok, signed_token} <- fetch_access_token(conn),
         {:ok, token}        <- verify_token(conn, signed_token, config),
         {user, metadata}    <- PersistentSessionCache.get(store_config, token) do

      {conn, user} = create(conn, user, config)

      conn =
        Conn.register_before_send(conn, fn conn ->
          CredentialsCache.delete(store_config, metadata[:access_token])
          PersistentSessionCache.delete(store_config, token)

          conn
        end)

      {conn, user}
    else
      _any -> {conn, nil}
    end
  end

  defp sign_token(conn, token, config) do
    Plug.sign_token(conn, signing_salt(), token, config)
  end

  defp signing_salt(), do: Atom.to_string(__MODULE__)

  defp fetch_access_token(conn) do
    case Conn.get_req_header(conn, "authorization") do
      [token | _rest] -> {:ok, token}
      _any            -> :error
    end
  end

  defp verify_token(conn, token, config),
    do: Plug.verify_token(conn, signing_salt(), token, config)

  defp store_config(config) do
    backend = Config.get(config, :cache_store_backend, Pow.Store.Backend.EtsCache)

    [backend: backend, pow_config: config]
  end
end
The above module includes renewal logic, and will return both an auth token and renewal token when a session is created.

  
    
  
  API authorization error handler


Create WEB_PATH/api_auth_error_handler.ex with the following:
defmodule MyAppWeb.APIAuthErrorHandler do
  use MyAppWeb, :controller
  alias Plug.Conn

  @spec call(Conn.t(), :not_authenticated) :: Conn.t()
  def call(conn, :not_authenticated) do
    conn
    |> put_status(401)
    |> json(%{error: %{code: 401, message: "Not authenticated"}})
  end
end
Now the protected routes will return a 401 error when an invalid token is used.

  
    
  
  Add API controllers


Create WEB_PATH/controllers/api/v1/registration_controller.ex:
defmodule MyAppWeb.API.V1.RegistrationController do
  use MyAppWeb, :controller

  alias Ecto.Changeset
  alias Plug.Conn
  alias MyAppWeb.ErrorHelpers

  @spec create(Conn.t(), map()) :: Conn.t()
  def create(conn, %{"user" => user_params}) do
    conn
    |> Pow.Plug.create_user(user_params)
    |> case do
      {:ok, _user, conn} ->
        json(conn, %{data: %{access_token: conn.private.api_access_token, renewal_token: conn.private.api_renewal_token}})

      {:error, changeset, conn} ->
        errors = Changeset.traverse_errors(changeset, &ErrorHelpers.translate_error/1)

        conn
        |> put_status(500)
        |> json(%{error: %{status: 500, message: "Couldn't create user", errors: errors}})
    end
  end
end
Create WEB_PATH/controllers/api/v1/session_controller.ex:
defmodule MyAppWeb.API.V1.SessionController do
  use MyAppWeb, :controller

  alias MyAppWeb.APIAuthPlug
  alias Plug.Conn

  @spec create(Conn.t(), map()) :: Conn.t()
  def create(conn, %{"user" => user_params}) do
    conn
    |> Pow.Plug.authenticate_user(user_params)
    |> case do
      {:ok, conn} ->
        json(conn, %{data: %{access_token: conn.private.api_access_token, renewal_token: conn.private.api_renewal_token}})

      {:error, conn} ->
        conn
        |> put_status(401)
        |> json(%{error: %{status: 401, message: "Invalid email or password"}})
    end
  end

  @spec renew(Conn.t(), map()) :: Conn.t()
  def renew(conn, _params) do
    config = Pow.Plug.fetch_config(conn)

    conn
    |> APIAuthPlug.renew(config)
    |> case do
      {conn, nil} ->
        conn
        |> put_status(401)
        |> json(%{error: %{status: 401, message: "Invalid token"}})

      {conn, _user} ->
        json(conn, %{data: %{access_token: conn.private.api_access_token, renewal_token: conn.private.api_renewal_token}})
    end
  end

  @spec delete(Conn.t(), map()) :: Conn.t()
  def delete(conn, _params) do
    conn
    |> Pow.Plug.delete()
    |> json(%{data: %{}})
  end
end
That's it!
You can now set up your client to connect to your API and generate session tokens. The session and renewal token should be send with the authorization header. When you receive a 401 error, you should renew the session with the renewal token and then try again.
You can run the following curl commands to test it out:
$ curl -X POST -d "user[email]=test@example.com&user[password]=secret1234&user[password_confirmation]=secret1234" http://localhost:4000/api/v1/registration
{"data":{"renewal_token":"RENEW_TOKEN","access_token":"AUTH_TOKEN"}}

$ curl -X POST -d "user[email]=test@example.com&user[password]=secret1234" http://localhost:4000/api/v1/session
{"data":{"renewal_token":"RENEW_TOKEN","access_token":"AUTH_TOKEN"}}

$ curl -X DELETE -H "Authorization: AUTH_TOKEN" http://localhost:4000/api/v1/session
{"data":{}}

$ curl -X POST -H "Authorization: RENEW_TOKEN" http://localhost:4000/api/v1/session/renew
{"data":{"renewal_token":"RENEW_TOKEN","access_token":"AUTH_TOKEN"}}


  
    
  
  OAuth 2.0


You may notice that the renew mechanism looks like refresh tokens in OAuth 2.0, and that's because the above setup is very similar since we use short lived session ids. In some cases it may make more sense to set up an OAuth 2.0 server rather than using the above setup.

  
    
  
  Test modules


# test/my_app_web/api_auth_plug_test.exs
defmodule MyAppWeb.APIAuthPlugTest do
  use MyAppWeb.ConnCase
  doctest MyAppWeb.APIAuthPlug

  alias MyAppWeb.{APIAuthPlug, Endpoint}
  alias MyApp.{Repo, Users.User}
  alias Plug.Conn

  @pow_config [otp_app: :my_app]

  setup %{conn: conn} do
    conn = %{conn | secret_key_base: Endpoint.config(:secret_key_base)}
    user = Repo.insert!(%User{id: 1, email: "test@example.com"})

    {:ok, conn: conn, user: user}
  end

  test "can create, fetch, renew, and delete session", %{conn: conn, user: user} do
    assert {_res_conn, nil} = run(APIAuthPlug.fetch(conn, @pow_config))

    assert {res_conn, ^user} = run(APIAuthPlug.create(conn, user, @pow_config))
    assert %{private: %{api_access_token: access_token, api_renewal_token: renewal_token}} = res_conn

    assert {_res_conn, nil} = run(APIAuthPlug.fetch(with_auth_header(conn, "invalid"), @pow_config))
    assert {_res_conn, ^user} = run(APIAuthPlug.fetch(with_auth_header(conn, access_token), @pow_config))
    assert {res_conn, ^user} = run(APIAuthPlug.renew(with_auth_header(conn, renewal_token), @pow_config))
    assert %{private: %{api_access_token: renewed_access_token, api_renewal_token: renewed_renewal_token}} = res_conn

    assert {_res_conn, nil} = run(APIAuthPlug.fetch(with_auth_header(conn, access_token), @pow_config))
    assert {_res_conn, nil} = run(APIAuthPlug.renew(with_auth_header(conn, renewal_token), @pow_config))
    assert {_res_conn, ^user} = run(APIAuthPlug.fetch(with_auth_header(conn, renewed_access_token), @pow_config))

    assert %Conn{} = run(APIAuthPlug.delete(with_auth_header(conn, "invalid"), @pow_config))
    assert {_res_conn, ^user} = run(APIAuthPlug.fetch(with_auth_header(conn, renewed_access_token), @pow_config))

    assert %Conn{} = run(APIAuthPlug.delete(with_auth_header(conn, renewed_access_token), @pow_config))
    assert {_res_conn, nil} = run(APIAuthPlug.fetch(with_auth_header(conn, renewed_access_token), @pow_config))
    assert {_res_conn, nil} = run(APIAuthPlug.renew(with_auth_header(conn, renewed_renewal_token), @pow_config))
  end

  defp run({conn, value}), do: {run(conn), value}
  defp run(conn), do: Conn.send_resp(conn, 200, "")

  defp with_auth_header(conn, token), do: Plug.Conn.put_req_header(conn, "authorization", token)
end
# test/my_app_web/controllers/api/v1/registration_controller_test.exs
defmodule MyAppWeb.API.V1.RegistrationControllerTest do
  use MyAppWeb.ConnCase

  @password "secret1234"

  describe "create/2" do
    @valid_params %{"user" => %{"email" => "test@example.com", "password" => @password, "password_confirmation" => @password}}
    @invalid_params %{"user" => %{"email" => "invalid", "password" => @password, "password_confirmation" => ""}}

    test "with valid params", %{conn: conn} do
      conn = post(conn, ~p"/registration", @valid_params)

      assert json = json_response(conn, 200)
      assert json["data"]["access_token"]
      assert json["data"]["renewal_token"]
    end

    test "with invalid params", %{conn: conn} do
      conn = post(conn, ~p"/registration", @invalid_params)

      assert json = json_response(conn, 500)
      assert json["error"]["message"] == "Couldn't create user"
      assert json["error"]["status"] == 500
      assert json["error"]["errors"]["password_confirmation"] == ["does not match confirmation"]
      assert json["error"]["errors"]["email"] == ["has invalid format"]
    end
  end
end
# test/my_app_web/controllers/api/v1/session_controller_test.exs
defmodule MyAppWeb.API.V1.SessionControllerTest do
  use MyAppWeb.ConnCase

  alias MyApp.{Repo, Users.User}

  @password "secret1234"

  setup do
    user =
      %User{}
      |> User.changeset(%{email: "test@example.com", password: @password, password_confirmation: @password})
      |> Repo.insert!()

    {:ok, user: user}
  end

  describe "create/2" do
    @valid_params %{"user" => %{"email" => "test@example.com", "password" => @password}}
    @invalid_params %{"user" => %{"email" => "test@example.com", "password" => "invalid"}}

    test "with valid params", %{conn: conn} do
      conn = post(conn, ~p"/api/v1/session", @valid_params)

      assert json = json_response(conn, 200)
      assert json["data"]["access_token"]
      assert json["data"]["renewal_token"]
    end

    test "with invalid params", %{conn: conn} do
      conn = post(conn, ~p"/api/v1/session", @invalid_params)

      assert json = json_response(conn, 401)
      assert json["error"]["message"] == "Invalid email or password"
      assert json["error"]["status"] == 401
    end
  end

  describe "renew/2" do
    setup %{conn: conn} do
      authed_conn = post(conn, ~p"/api/v1/session", @valid_params)

      {:ok, renewal_token: authed_conn.private[:api_renewal_token]}
    end

    test "with valid authorization header", %{conn: conn, renewal_token: token} do
      conn =
        conn
        |> Plug.Conn.put_req_header("authorization", token)
        |> post(~p"/api/v1/session/renew")

      assert json = json_response(conn, 200)
      assert json["data"]["access_token"]
      assert json["data"]["renewal_token"]
    end

    test "with invalid authorization header", %{conn: conn} do
      conn =
        conn
        |> Plug.Conn.put_req_header("authorization", "invalid")
        |> post(~p"/api/v1/session/renew")

      assert json = json_response(conn, 401)
      assert json["error"]["message"] == "Invalid token"
      assert json["error"]["status"] == 401
    end
  end

  describe "delete/2" do
    setup %{conn: conn} do
      authed_conn = post(conn, ~p"/api/v1/session/", @valid_params)

      {:ok, access_token: authed_conn.private[:api_access_token]}
    end

    test "invalidates", %{conn: conn, access_token: token} do
      conn =
        conn
        |> Plug.Conn.put_req_header("authorization", token)
        |> delete( ~p"/api/v1/session/")

      assert json = json_response(conn, 200)
      assert json["data"] == %{}
    end
  end
end

Used to set up context API.

Used to set up extensions to enable parts of extension for auto-discovery.
This exists to prevent unnecessary Code.ensure_compiled/1 calls, and will
let the extension define what modules it has.

  
    
  
  Usage


defmodule MyCustomExtension do
  use Pow.Extension.Base

  @impl true
  def ecto_schema?(), do: true
end

Operation functions that glues operation calls to context module.
A custom context module can be used instead of the default Pow.Ecto.Context
if a :users_context key is passed in the configuration.

Module to generate UUID binary.
The token generation code is pulled from
https://github.com/elixir-ecto/ecto/blob/v2.2.10/lib/ecto/uuid.ex#L1, and is
under Apache 2 license.

Plug helper functions.

Pow.Plug.Base behaviour
    



      
This plug macro will set :pow_config as private, and attempt to fetch and
assign a user in the connection if it has not already been assigned. The user
will be assigned automatically in any of the operations.
Any writes to backend store or client should occur in :before_send callback
as defined in Plug.Conn. To ensure that the callbacks are called in the
order they were set, a register_before_send/2 function is used to set
callbacks instead of Plug.Conn.register_before_send/2.

  
    
  
  Configuration options


	:credentials_cache_store - the credentials cache store. This value defaults to
{Pow.Store.CredentialsCache, backend: Pow.Store.Backend.EtsCache}. The
Pow.Store.Backend.EtsCache backend store can be changed with the
:cache_store_backend option.

	:cache_store_backend - the backend cache store. This value defaults to
Pow.Store.Backend.EtsCache.



  
    
  
  Example


defmodule MyAppWeb.Pow.CustomPlug do
  use Pow.Plug.Base

  @impl true
  def fetch(conn, _config) do
    user = fetch_user_from_cookie(conn)

    {conn, user}
  end

  @impl true
  def create(conn, user, _config) do
    conn = update_cookie(conn, user)

    {conn, user}
  end

  @impl true
  def delete(conn, _config) do
    delete_cookie(conn)
  end
end

This module can sign and verify messages.
Based on Phoenix.Token.

Pow.Plug.RequireAuthenticated 
    



      
This plug ensures that a user has been authenticated.
You can see Pow.Phoenix.PlugErrorHandler for an example of the error
handler module.

  
    
  
  Example


plug Pow.Plug.RequireAuthenticated,
  error_handler: MyApp.CustomErrorHandler

This plug ensures that a user hasn't already been authenticated.
You can see Pow.Phoenix.PlugErrorHandler for an example of the error
handler module.

  
    
  
  Example


plug Pow.Plug.RequireNotAuthenticated,
  error_handler: MyAppWeb.Pow.ErrorHandler

This plug will handle user authorization using session.
The plug will store user and session metadata in the cache store backend. The
session metadata has at least an :inserted_at and a :fingerprint key. The
:inserted_at value is used to determine if the session has to be renewed,
and is set each time a session is created. The :fingerprint will be a random
unique id and will stay the same if a session is renewed.
When a session is renewed the old session is deleted and a new created.
You can add additional metadata to sessions by setting or updated the
assigned private :pow_session_metadata key in the conn. The value has to be
a keyword list.
The session id used in the client is signed using Pow.Plug.sign_token/4 to
prevent timing attacks.

  
    
  
  Example


@pow_config [
  repo: MyApp.Repo,
  user: MyApp.User,
  current_user_assigns_key: :current_user,
  session_key: "auth",
  credentials_cache_store: {Pow.Store.CredentialsCache,
                            ttl: :timer.minutes(30),
                            namespace: "credentials"},
  session_ttl_renewal: :timer.minutes(15),
  cache_store_backend: Pow.Store.Backend.EtsCache,
  users_context: Pow.Ecto.Users
]

# ...

plug Plug.Session, @session_options
plug Pow.Plug.Session, @pow_config

  
    
  
  Configuration options


	:credentials_cache_store - see Pow.Plug.Base.

	:cache_store_backend - see Pow.Plug.Base.

	:session_key - session key name, defaults to "auth". If :otp_app is
used it'll automatically prepend the key with the :otp_app value.

	:session_ttl_renewal - the ttl in milliseconds to trigger renewal of
sessions. Defaults to 15 minutes in miliseconds.



  
    
  
  Custom metadata


The assigned private :pow_session_metadata key in the conn can be populated
with custom metadata. This data will be stored in the session metadata when
the session is created, and fetched in subsequent requests.
Here's an example of how one could add sign in timestamp, IP, and user agent
information to the session metadata:
def append_to_session_metadata(conn) do
  client_ip  = to_string(:inet_parse.ntoa(conn.remote_ip))
  user_agent = get_req_header(conn, "user-agent")

  metadata =
    conn.private
    |> Map.get(:pow_session_metadata, [])
    |> Keyword.put_new(:first_seen_at, DateTime.utc_now())
    |> Keyword.put(:ip, client_ip)
    |> Keyword.put(:user_agent, user_agent)

  Plug.Conn.put_private(conn, :pow_session_metadata, metadata)
end
The :first_seen_at will only be set if it doesn't already exist in the
session metadata, while :ip and :user_agent will be updated each time the
session is created.
The function should be called after Pow.Plug.Session.call/2 has been called
to ensure that the metadata, if any, has been fetched.

  
    
  
  Session expiration


Pow.Store.CredentialsCache will, by default, invalidate any session token
30 minutes after it has been generated. To keep sessions alive the
:session_ttl_renewal option is used to determine when a session token
becomes stale and a new session ID has to be generated for the user (deleting
the previous one in the process).
If :session_ttl_renewal is set to zero, a new session token will be
generated on every request.
To change the amount of time a session can be alive, both the TTL for
Pow.Store.CredentialsCache and :session_ttl_renewal option should be
changed:
plug Pow.Plug.Session, otp_app: :my_app,
  session_ttl_renewal: :timer.minutes(1),
  credentials_cache_store: {Pow.Store.CredentialsCache, ttl: :timer.minutes(15)}
In the above, a new session token will be generated when a request occurs
more than a minute after the current session token was generated. The
session is invalidated if there is no request for the next 14 minutes.
There are no absolute session timeout; sessions can be kept alive
indefinitely.

Handles pow users context for user.

  
    
  
  Usage


This module will be used by pow by default. If you
wish to have control over context functions, you can
do configure lib/my_project/users/users.ex
the following way:
defmodule MyApp.Users do
  use Pow.Ecto.Context,
    repo: MyApp.Repo,
    user: MyApp.Users.User

  def create(params) do
    pow_create(params)
  end
end
Remember to update configuration with users_context: MyApp.Users.
The following Pow functions can be accessed:
	pow_authenticate/1
	pow_create/1
	pow_update/2
	pow_delete/1
	pow_get_by/1


  
    
  
  Configuration options


	:repo - the ecto repo module (required)
	:user - the user schema module (required)
	:repo_opts - keyword list options for the repo, :prefix can be set here


      


      
        Summary


  
    Types
  


    
      
        
  
    
    Pow.Ecto.Schema - Pow v1.0.39
    
    

    


  
  

    
Pow.Ecto.Schema behaviour
    



      
Handles the Ecto schema for user.
The macro will create a @pow_fields module attribute, and append fields to
it using the attributes from Pow.Ecto.Schema.Fields.attrs/1. Likewise, a
@pow_assocs module attribute is also generated for associations. The
pow_user_fields/0 macro will use these attributes to create fields and
associations in the ecto schema.
The macro will add two overridable functions to your module; changeset/2
and verify_password/2. See the customization section below for more.
The following helper functions are added for changeset customization:
	pow_changeset/2,
	pow_verify_password/2
	pow_user_id_field_changeset/2
	pow_password_changeset/2,
	pow_current_password_changeset/2,

Finally pow_user_id_field/0 function is added to the module that is used to
fetch the user id field name.
A @pow_config module attribute is created containing the options that were
passed to the macro with the use Pow.Ecto.Schema, ... call.
See Pow.Ecto.Schema.Changeset for more.

  
    
  
  Usage


Configure lib/my_project/users/user.ex the following way:
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema,
    user_id_field: :email,
    password_hash_verify: {&Pow.Ecto.Schema.Password.pbkdf2_hash/1,
                            &Pow.Ecto.Schema.Password.pbkdf2_verify/2},
    password_min_length: 8,
    password_max_length: 4096

  schema "users" do
    field :custom_field, :string

    pow_user_fields()

    timestamps()
  end

  def changeset(user_or_changeset, attrs) do
    pow_changeset(user_or_changeset, attrs)
  end
end
Remember to add user: MyApp.Users.User to your configuration.

  
    
  
  Configuration options


	:user_id_field - the field to use for user id. This value defaults to
:email, and the changeset will automatically validate it as an e-mail.


  
    
  
  Customize Pow fields


Pow fields can be overridden if the field name and type matches:
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema

  schema "users" do
    field :encrypted_password, :string
    field :password_hash, :string, source: :encrypted_password

    pow_user_fields()

    timestamps()
  end
end
The same holds true for associations:
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema

  @pow_assocs {:belongs_to, :invited_by, __MODULE__, []}
  @pow_assocs {:has_many, :invited, __MODULE__, []}

  schema "users" do
    belongs_to :invited_by, __MODULE__, foreign_key: :user_id

    pow_user_fields()

    timestamps()
  end
end
An @after_compile callback will emit IO warning if there are missing fields
or associations. You can forego the pow_user_fields/0 call, and write out
the whole schema instead:
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema, user_id_field: :email

  schema "users" do
    field :email,            :string
    field :password_hash,    :string
    field :current_password, :string, virtual: true
    field :password,         :string, virtual: true
    field :confirm_password, :string, virtual: true

    timestamps()
  end
end
If you would like these warnings to be raised during compilation you can add
elixirc_options: [warnings_as_errors: true] to the project options in
mix.exs.
The warning is also emitted if the field has an invalid primitive Ecto type.
It'll not be emitted for custom Ecto types.

  
    
  
  Customize Pow changeset


You can extract individual changeset functions to modify the changeset flow
entirely. As an example, this is how you can remove the validation check for
confirm password in the changeset function:
defmodule MyApp.Users.User do
  use Ecto.Schema
  use Pow.Ecto.Schema

  import Pow.Ecto.Schema.Changeset, only: [new_password_changeset: 3]

  # ...

  def changeset(user_or_changeset, attrs) do
    user_or_changeset
    |> pow_user_id_field_changeset(attrs)
    |> pow_current_password_changeset(attrs)
    |> new_password_changeset(attrs, @pow_config)
  end
end
Note that the changeset functions in Pow.Ecto.Schema.Changeset require the
Pow ecto module configuration that is passed to the
use Pow.Ecto.Schema, ... call. This can be fetched by using the
@pow_config module attribute.

Pow.Ecto.Schema.Changeset 
    



      
Handles changesets functions for Pow schema.
These functions should never be called directly, but instead the functions
build in macros in Pow.Ecto.Schema should be used. This is to ensure
that only compile time configuration is used.
Pow.Ecto.Schema.Password is by default used to hash and verify passwords.

  
    
  
  Configuration options


	:password_min_length   - minimum password length, defaults to 8

	:password_max_length   - maximum password length, defaults to 4096

	:password_hash_verify  - the password hash and verify anonymous
functions or MFAs, defaults to:
{&Pow.Ecto.Schema.Password.pbkdf2_hash/1,
&Pow.Ecto.Schema.Password.pbkdf2_verify/2}
It may be anonymous functions of MFAs.

	:email_validator       - the email validation anonymous function or
MFA, defaults to:
&Pow.Ecto.Schema.Changeset.validate_email/1
  The function should either return :ok, :error, or
  {:error, reason}.

Pow.Ecto.Schema.Fields 
    



      
Handles the Ecto schema fields for user.

Pow.Ecto.Schema.Migration 
    



      
Generates schema migration content.

  
    
  
  Configuration options


	:repo - the ecto repo to use. This value defaults to the derrived
context base repo from the context_base argument in gen/2.

	:table - the ecto table name, defaults to "users".

	:attrs - list of attributes, defaults to the results from
Pow.Ecto.Schema.Fields.attrs/1.

	:indexes - list of indexes, defaults to the results from
Pow.Ecto.Schema.Fields.indexes/1.



      


      
     